

An Introduction to

# Scilab

from a Matlab User's Point of View

Version 2.6-1.0

Eike Rietsch

Copyright ©2001, 2002 by Eike Rietsch

Permission is granted to anyone to make or distribute verbatim copies of this document as received, in any medium, provided that the copyright notice and permission notice are preserved, and that the distributor grants the recipient permission for further redistribution as permitted by this notice.

IBM® and RS/6000® are registered trademarks of IBM Corp.

Macsyma™ is a trademark of Macsyma Inc.

Maple™ is a trademark of Waterloo Maple Inc.

Matlab™ is a trademark of The Mathworks, Inc.

Mathematica™ is a trademark of Wolfram Research, Inc.

Microsoft™ and Microsoft Windows™ are trademarks of Microsoft Corp.

Sun™ and Solaris™ are trademarks of Sun Microsystems, Inc.

UNIX® is a registered trademark of The Open Group.

Scilab© is copyrighted by INRIA, France

*To Antje*



# Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Preliminaries</b>	<b>4</b>
2.1	Customizing Scilab for Windows . . . . .	4
2.1.1	Startup File . . . . .	4
2.1.2	Fonts . . . . .	4
2.1.3	Paging . . . . .	4
2.1.4	Copy and Paste . . . . .	5
2.2	Interruption/Termination of Scripts and Scilab Session . . . . .	5
2.3	Help . . . . .	5
2.4	Emulated Matlab functions . . . . .	6
<b>3</b>	<b>Syntax</b>	<b>7</b>
3.1	Arithmetic Statements . . . . .	7
3.2	Built-in Constants . . . . .	10
3.3	Comparison Operators . . . . .	11
3.4	Flow Control . . . . .	12
3.5	General Functions . . . . .	15
<b>4</b>	<b>Variable Types</b>	<b>19</b>
4.1	Numeric Variables — Scalars and Matrices . . . . .	21
4.2	Special Matrices . . . . .	24
4.3	Character String Variables . . . . .	25
4.3.1	Creation and Manipulation of Character Strings . . . . .	25
4.3.2	Strings with Scilab Expressions . . . . .	32
4.3.3	Symbolic String Manipulation . . . . .	34
4.4	Boolean Variables . . . . .	35
4.5	Lists . . . . .	39
4.5.1	Ordinary lists (list) . . . . .	39
4.5.2	Typed lists (tlist) . . . . .	43
4.5.3	Matrix-oriented typed lists (mlist) . . . . .	49
4.6	Polynomials . . . . .	50

<b>5</b>	<b>Functions</b>	<b>54</b>
5.1	General . . . . .	54
5.2	Functions that Operate on Scalars and Matrices . . . . .	56
5.2.1	Basic Functions . . . . .	56
5.2.2	Elementary Mathematical Functions . . . . .	59
5.2.3	Special Functions . . . . .	62
5.2.4	Linear Algebra . . . . .	63
5.2.5	Signal-Processing Functions . . . . .	64
5.3	File Input and Output . . . . .	66
5.3.1	Opening and Closing of Files . . . . .	66
5.3.2	Functions <code>mgetl</code> and <code>mputl</code> . . . . .	67
5.3.3	Functions <code>read</code> and <code>write</code> . . . . .	70
5.3.4	Functions <code>load</code> and <code>save</code> . . . . .	73
5.3.5	Functions <code>mput</code> and <code>mget/mgeti</code> . . . . .	74
5.3.6	Functions <code>input</code> and <code>disp</code> . . . . .	75
5.3.7	Function <code>xgetfile</code> . . . . .	75
5.4	Utility Functions . . . . .	76
<b>6</b>	<b>Scripts</b>	<b>78</b>
<b>7</b>	<b>User Functions</b>	<b>84</b>
7.1	Functions in Files . . . . .	90
7.2	In-line Functions . . . . .	91
7.3	Functions for operator overloading . . . . .	92
7.4	Translation of Matlab-4 m-files to Scilab Format . . . . .	97
<b>8</b>	<b>Function Libraries</b>	<b>98</b>
<b>9</b>	<b>Gotchas</b>	<b>101</b>
<b>A</b>	<b>Matlab functions and their Scilab Equivalents</b>	<b>102</b>

# List of Tables

3.1	List of arithmetic operators . . . . .	8
3.2	Built-in constants . . . . .	10
3.3	General functions . . . . .	16
4.1	Variable types . . . . .	19
4.2	Utility functions for managing/classifying of variables . . . . .	20
4.3	Integer conversion functions . . . . .	22
4.4	Special matrices . . . . .	24
4.5	Functions that manipulate strings . . . . .	26
4.6	Functions that operate on, or output, boolean variables . . . . .	38
4.7	Functions that create, or operate on, lists . . . . .	40
4.8	Functions related to polynomials and rational functions . . . . .	51
5.1	Basic arithmetic functions . . . . .	57
5.2	Elementary transcendental functions . . . . .	60
5.3	Matrix functions . . . . .	61
5.4	Special functions . . . . .	62
5.5	Functions for sparse matrices . . . . .	62
5.6	Linear algebra . . . . .	63
5.7	Functions for signal processing . . . . .	64
5.8	Functions that open, query, manipulate, and close files . . . . .	67
5.9	Functions that input data from files or keyboard . . . . .	68
5.10	Functions that output data to files or to the Scilab window . . . . .	69
5.11	Utility functions . . . . .	76
7.1	Functions/commands/keywords relevant for user functions . . . . .	85
7.2	Operator codes used to construct function names for operator overloading . . . . .	94





# Chapter 1

## Introduction

For almost 10 years I have been a heavy user of Matlab; this manual is the result of my effort to learn Scilab. Consequently, it is written from the point of view of someone who is familiar with Matlab and wants to use this knowledge to ease his entry into Scilab. Thus features that are the same in both systems are “glossed over” to some degree and more space is devoted to those features where the two differ. As a result, this manual is not really suited for someone who is not familiar with either Matlab or Scilab (unless he is desperate or brilliant). Documentation more suitable for a novice is available on-line or in bookstores.

Initially, I planned to organize the material ordered by Matlab functions since this was the way I approached the problem of converting Matlab functions to Scilab. However, there is not always an exact correspondence between Matlab and Scilab functions and syntax; furthermore, Scilab has features not available in Matlab, and so I reconsidered. Hence, this manual explains Scilab’s functionality by drawing on the experience and expectations of a Matlab user.

To aid in the conversion of Matlab functions the table in Appendix A lists Matlab functions and their functional equivalents. Furthermore, there are three separate indexes: a general index, an index of Scilab functions and an index of Matlab functions. So one can look up quite a number of Matlab function to find out what means there are to achieve the same end in Scilab. A user trying to figure out how to implement, say, a Matlab structure will be directed to Scilab lists. Someone who wants to understand the difference in the definition of workspace—which has the potential to trip up the unsuspecting—will need to look in the general index which points to those pages that describe this difference.

Incidentally, there is a subdirectory of a subdirectory in the Scilab directory with Scilab functions that “emulate” Matlab functions. As explained more fully in Section 2.4 I do not advocate their use. Using such emulations deprives the user of the flexibility and power Scilab offers. In most cases it is a concept one needs to emulate not a function.

This manual is organized in a number of chapters, sections, and subsections. Obviously, this is arbitrary and reflects my own choices. Several sections have tables of functions or operators pertinent to the subject matter discussed. Due to some overlap one and the same function may show up in several different tables.

It was tempting to use unadulterated screen dumps as examples. However Scilab wastes screen real estate the same way *format loose* does in Matlab — except, in Scilab, there is no equivalent to *format tight*, which suppresses the extra line-feeds. Hence, to conserve space, most examples are reproduced without some of these extra empty lines.

In compiling this manual I used Scilab 2.6 and the standard Scilab documentation.

*Introduction To Scilab - Users Guide* by the Scilab Group

*Une Introduction à Scilab* by Bruno Pinçon

*Scilab Bag of Tricks* by Lydia E. van Dijk and Christoph L. Spiel

All three can be downloaded from the INRIA web site (<http://www-rocq.inria.fr/scilab/>), which also has manuals in languages other than English and French. I also drew freely on newsgroup discussions ([comp.soft-sys.math.scilab](mailto:comp.soft-sys.math.scilab)), in particular contributions by Bruno Pinçon, Alexander Vigodner, Enrico Segre, Lydia van Dijk, and Helmut Jarausch.

From newsgroup discussions I got the impression that most users run Scilab on Unix (particularly Linux) machines. I, on the other hand, use Matlab and Scilab on Windows PC's. I do have a Scilab installation on a Sun workstation running Solaris, but use it only occasionally for quick calculations in a Unix environment. While I do not expect significant differences in the use of Scilab on different platforms, this pattern of use does color this manual. However, I am not completely Windows-centric: affected by many years of Unix use, I tend to favor the Unix term “directory” over the PC term “folder”.

Every now and then this manual contains judgements regarding the relative merits of features in Matlab and Scilab. They represent my personal biases, experiences, and — presumably — a lack of knowledge as well.

Obviously, I cannot claim to cover all Matlab functions or Scilab functions. The selection is largely based on the subset of functions and syntactical features that I have been using over the years. But among all the omissions one is glaring. I do not discuss plotting. Were I unaware of Matlab, I would consider Scilab's plotting facility superb. But now I am spoiled. However, I understand that a new object-oriented plot library is under development, and I am looking forward to its release. Furthermore, plotting is such an important and extensive subject that it deserves a manual of its own (as is the case for Matlab).

Finally, the typographic conventions used are:

**Red typewriter font is used for Scilab commands, functions, variables, ...**

**Blue slanted typewriter font is used for Matlab commands, functions, variables, ...**

Black typewriter font is used for general operating system-related terms and filenames outside of code fragments.

Keyboard keys, such as the Return key, are written with the name enclosed in angle brackets: <RETURN>. In the section on operator overloading angle brackets are also used to enclose operand types and operator codes.

## Acknowledgment

Special thanks go to Glenn Fulford who was kind enough to review this manuscript and offer

suggestions and critique and, in particular, to Serge Steer who not only provided a list of corrections but also an extensive compilation of the differences between Scilab and Matlab; I used for my own education and included what I learned.

# Chapter 2

## Preliminaries

### 2.1 Customizing Scilab for Windows

#### 2.1.1 Startup File

Commands that should be executed at the beginning of a Scilab Session can be put in the startup file `.scilab` (the dot “.” as the first character of the file name betrays the Unix heritage of Scilab). On a PC running Windows this start-up file must be in directory `bin` of the Scilab directory.

#### 2.1.2 Fonts

Screen fonts can be set in two different ways. Either click on the Edit Button and then on **Choose Font** in the drop-down menu. Alternatively, click the right mouse button in the Scilab window and select **Choose Font**. To save the selected font, click the right mouse button in the Scilab window and select Update `scilab.ini`.

The Help Window comes with a proportional font preselected. However, in general a fixed-width font produces a more readable display, in particular with matrices. The fonts in the Help Window can be set by clicking the Format Button of the Help Window and selecting Font in the drop-down menu. In this case the selected font is saved automatically.

#### 2.1.3 Paging

By default, display of a long array or vector is halted after a number of lines have been printed to the screen, and the message `[More (y or n) ?]` is displayed. The number of lines displayed can be controlled via the `lines` command. Paging (and that message) can be suppressed by means of the command `lines(0)`. If this appears desirable, this command can be put in the startup file `.scilab` to be run at start-up.

### 2.1.4 Copy and Paste

The standard Windows keyboard shortcuts for **Copy** and **Paste** do not work in the Scilab window (they do work in the **Help** window). However, the drop-down menu of the Scilab window's Edit Button has **Copy to Clipboard** and **Paste** commands. The same commands can also be found in the menu that opens up when one clicks the right mouse button in the Scilab window.

## 2.2 Interruption/Termination of Scripts and Scilab Session

Scilab has a feature that is sorely missed in Matlab: a reliable facility to interrupt or terminate a running program. The command **abort** allows one to terminate execution of a function or script, e. g. in debugging mode after a **pause** has been executed and continuation of the execution is not desired. In Matlab the usual way to achieve this goal is to clear all variables and thus to force a fatal error with the **return** command — and even this does not work every time. The **abort** command can also be invoked from the Control Menu and does what it says: it aborts any running program. A less drastic intervention is **Interrupt**, also available from the Control Menu. It interrupts a running program to allow keyboard entry (note that a program interruption in Scilab creates a new workspace; what this means is explained on page 15). Execution of the program can be resumed by typing **resume** or **return**. The same objective can be achieved by means of the **Resume** menu item in the Control Menu or its keyboard shortcut <Alt c> followed by <Alt e> (press down the <Alt> key and hit <c> and then <e>). There are keyboard shortcuts for all commands in this menu.

The commands **quit** and **exit** can be used to terminate a Scilab session. Both commands exist in Matlab as well, and **exit** behaves like its Matlab counterpart. The **quit** command is somewhat different. If called from a higher workspace level it reduces the level by 2 (see the discussion of **pause** on page 15). If called from level 0 it terminates Scilab. In this case **quit** also executes a termination script, **scilab.quit**, located in the Scilab root directory (on a PC running Windows something like C:\Program Files\Scilab). This script can be modified by the user and is comparable to **finish** in Matlab. Of course, one can also terminate Scilab by clicking on **Exit** in the **File** menu or the **close** box in the right upper corner of the Scilab window.

## 2.3 Help

The help facility is similar to Matlab's. Typing **help sin**, for example, brings up a separate help window with information about **sin**. Typing **help symbols** brings up a table of symbols and the corresponding alphabetic equivalent to be used in the **help** command. For example, to find out what **.\*** does type **help star**. Unfortunately, in some instances, one has to type in misspelled words such as “tilda” (for “tilde”) or “semicolon” (for “semicolon”).

The command **apropos**, somewhat equivalent to Matlab's **lookfor**, performs a string search and lists the result in a separate window. Selecting a command in this list and clicking on **OK** brings

up the help window for that command. Both, `help` and `apropos`, can also be invoked from the Help Menu on the menu bar (menu items Topic and Apropos). The third item, Help Dialog, on the Help Button’s Drop-Down Menu opens a window with two sections. One of them lists some 25 topics such as “Input/Output Functions”, “Linear Algebra”, “Character String Manipulation”, etc. Clicking on a topic brings up, in the second section, a one-line-per-function list of relevant Scilab functions—a nice help to get started. It is particularly convenient that selecting a function and clicking the “Show” button opens a window with the help file for this function.

## 2.4 Emulated Matlab functions

As already mentioned in the Introduction, the Scilab distribution comes with a directory, `SCIDIR\macros\mtlb`, where `SCIDIR` denotes the Scilab root directory (in Windows something like `C:\Program Files\Scilab`). In this directory there are some 80 function that “emulate” Matlab functions; only four of them have help files. For several reasons I do not advocate their use. First of all, this kind of “translation” of a Matlab object to Scilab may prevent a user from fully exploiting powerful features Scilab offers. An example is `mtlb_cell` (most of the functions in the directory `mtlb` start with the prefix `mtlb_`), which emulates the Matlab function `cell` by means of a typed list. But there are many different ways a Matlab cell array can be expressed in Scilab. If all cell entries are strings then a string matrix is the appropriate “translation” in Scilab. Using `mtlb_cell` instead deprives one of the benefits string matrices offer (such as the overloaded `+` operator and the functionality of `length`). In other situations a ordinary list or a list of lists may be more appropriate.

A second reason for not using the functions in this directory is that careless use may lead to wrong results. An example is Matlab function `diff` and its Scilab “emulation” `mtlb_diff`. With only one argument these two functions produce the same result; but with two arguments this is not necessarily the case since the second argument in `diff` serves a different purpose than in `mtlb_diff`. For example, in Matlab

```
>> diff([1:10].^2,2)
ans =
     2     2     2     2     2     2     2     2
```

whereas in Scilab

```
-->mtlb_diff([1:10].^2,2)
ans =
!   8.   12.   16.   20.   24.   28.   32.   36. !
```

This just illustrates why it is better to stay away from these functions—at most use them as suggestions for implementing something in Scilab.

# Chapter 3

## Syntax

### 3.1 Arithmetic Statements

Scilab syntax is generally quite like Matlab syntax. This means that someone familiar with Matlab knows how to write basic Scilab commands such as

```
// These are simple examples
-->a = 3; b = 7.2;

-->c = a + b^2 - sin(3.1415926/2)
c =
  53.84
```

As shown in this example the Scilab prompt is `-->`, and any statements following it represent user input. Comments are indicated by two slashes (`//`): everything to the right of the slashes is ignored by the interpreter/compiler. Like in Matlab, several statements can be on one line as long as they are separated by commas or semicolons. Semicolons suppress the display of results, commas do not.

Names of Scilab variables and functions must begin with a letter or one of the following special characters `%`, `#`, `!`, `$`, `?`, and the underscore `_`. Subsequent characters may be alphanumeric or the special characters `#`, `!`, `$`, `?`, and `_`. Thus `%` is only allowed as the first character of a variable name. Variables starting with `%` generally represent built-in constants or functions that overload operators. Variable names may be of arbitrary length, but all except the first 24 characters are disregarded (Matlab uses the first 31 characters).

```
-->a123456789012345678901234567890123456789012345678901234567890 = 34
a12345678901234567890123 =
  34.
```

Variable names are case-sensitive (i. e. Scilab distinguishes between upper-case and lower-case letters). A semicolon (;) terminating a statement indicates that the result should not be displayed whereas a comma or a <RETURN> prompts a display of the result.

To create expressions, Scilab uses the same basic arithmetic operators Matlab does, but with two options for exponentiation.

+	Addition
-	Subtraction
*	Matrix multiplication
.*	Array multiplication
.*. .	Kronecker multiplication
/	Division
\	Left matrix division
./	Array division
.\	Left array division
./.	Kronecker division
.\.	Kronecker left division
^ or **	Matrix exponentiation
.^	Array exponentiation
/'	Matrix complex transposition
./'	Array transposition

Table 3.1: List of arithmetic operators

Statements can continue over several lines. Similar to Matlab's syntax, continuation of a statement is indicated by two or more dots, .. (Matlab requires at least three dots).

Numbers can be used with and without decimal point. Thus the numbers 1, 1., and 1.0 are equivalent. However, in both Scilab and Matlab, the decimal points does double duty. In conjunction with the operators \*, /, and ^ it indicates that operations on vectors and matrices are to performed on an element-by-element basis. This leads to ambiguities that can cause problems for the unsuspecting.

```
-->x = 1/[1 2 3]      1a
x =
!      .0714286 !
!      .1428571 !
!      .2142857 !
```

```
-->x = 1./[1 2 3]    1b
x =
!      .0714286 !
!      .1428571 !
```



```

!      .2142857 !

-->x = 1 ./[1 2 3]      1c
x =
!      1.      .5      .3333333 !

```

Statements `1b` and `1c` look very similar and yet they produce quite different results. The reason for the difference is the space between the zero and the dot in `1c` where the dot is interpreted as part of the operator `./` whereas in `1b` it is interpreted as the decimal point of the number 1. In Matlab, on the other hand, statements `1b` and `1c` produce the same result (the one produced in Scilab by `1c`), and `1a` causes an error message. In Scilab, on the other hand, `a` is a solution of `a'*[1 2 3]' = 1`. More general, if

$$x^T A^T = b^T \quad (3.1)$$

where the superscripted  $T$  denotes transposition, then `x = b'/A'` computes the unknown  $x^T$ . Since Eq. 3.1 is equivalent to

$$Ax = b \quad (3.2)$$

$x$  can also be computed from `x = A\b`. Hence, `(b'/A)'` is equivalent to `A\b`. The latter is also Matlab syntax. Thus

```

-->x = [1 2 3]'\ 1
x =
!      .0714286      .1428571      .2142857 !

```

In addition to single-variable assignments, Scilab has tuple assignments in which multiple variables are assigned values in a single statement. An example is

```

--> [u,v,w] = (1,2,3)
w =
3.
v =
2.
u =
1.

```

Note that the commas on the right-hand and the left-hand side are required; they cannot be replaced by blanks). This construct bears some similarity with Matlab's `deal` function, but it is less powerful. For example, the number of Scilab objects on the right-hand side must equal that on the left hand side. Even if one wanted to assign the same value to all three variables one would still have to write it out three times; thus `[u,v,w] = (1)` is not allowed.

A feature peculiar to Scilab is the order (from right to left) in which variables in a left-hand bracketed expression are displayed; as shown in the example above the rightmost variable, `w`, is displayed first, followed by `u` and, finally, `v`.

A handy use of the tuple assignment is swapping of values. With variables `u` and `v` defined above

```
-->[u,v] = (v,u)
v =
    1.
u =
    2.
```

## 3.2 Built-in Constants

Scilab	Matlab	Description
<code>%i</code>	<code>i, j</code>	Imaginary unit ( $\sqrt{-1}$ )
<code>%e</code>	<code>e</code>	Euler's constant ( $e = 2.7182818 \dots$ )
<code>%pi</code>	<code>pi</code>	Ratio of circumference to diameter of a circle; ( $\pi = 3.1415926 \dots$ )
<code>%eps</code>	<code>eps</code>	Machine $\epsilon$ ( $\approx 2.2 \cdot 10^{-16}$ ); smallest number such that $1 + \epsilon > 1$
<code>%inf</code>	<code>inf</code>	Infinity ( $\infty$ )
<code>%nan</code>	<code>NaN</code>	Not a number
<code>%s</code>		Polynomial <code>s=poly(0,'s')</code>
<code>%z</code>		Polynomial <code>z=poly(0,'z')</code>
<code>%t</code>	<code>logical(1)</code>	Boolean variable: logical true
<code>%f</code>	<code>logical(0)</code>	Boolean variable: logical false
<code>%io</code>		Two-element vector with file identifiers for standard I/O

Table 3.2: Built-in constants

Table 3.2 lists Scilab's special, built-in constants together with their Matlab equivalents (where they exist). Unlike constants in Matlab they are protected and cannot be overwritten. This has benefits; in Matlab a variable such as `i` can be overwritten inadvertently if it is redefined by, for example, its use as an index.

In many respects, keyboard (standard input) and Scilab window (standard output) are treated like files, and `%io(1)` (usually 5) is the file identifier for the keyboard and `%io(2)` (usually 6) is the file identifier for the Scilab window.

### 3.3 Comparison Operators

Scilab uses the same comparison operators Matlab does, but with two choices for the “not equal” operator.

<	less than
>	greater than
<=	less than or equal to
>=	greater than or equal to
==	equal to
<> or ~=	not equal to

The result of a valid expression involving any of these operators — such as  $a > 0$  — is a boolean variable (`%t` or `%f`) or a matrix of boolean variables. These boolean variables are discussed later in section 4.4.

In Scilab the first four operators are only defined for real numbers; in Matlab complex numbers are allowed but only the real part is used for the comparison.

The last two operators compare objects. Examples are

```
--> [1 2 3] == 1      2a
ans =
! T F F !
```

```
--> [1 2 3] == [1 2]  2b
ans =
F
```

```
--> [1 2] == ['1','2'] 2c
ans =
F
```

In Matlab `2a` produces the same result, `2b` aborts with an error message, and `2c` creates the boolean vector `[0 0]`.

### 3.4 Flow Control

Scilab's flow control syntax mirrors that used by Matlab.

Scilab	Matlab	
<code>break</code>	<code>break</code>	Force exit from a loop
<code>case</code>	<code>case</code>	Start clause within a <code>select</code> block
<code>elseif</code>	<code>elseif</code>	Start a conditional alternative in an <code>if</code> block
<code>else</code>	<code>else/otherwise</code>	Start the alternative in an <code>if</code> or <code>select</code> block
<code>end</code>	<code>end</code>	Terminate <code>for</code> , <code>if</code> , <code>select</code> , and <code>while</code> blocks
<code>errcatch</code>	<code>try/catch</code>	Traps error and with several possible actions
<code>for</code>	<code>for</code>	Start a loop with a generally known number of repetitions
<code>if</code>	<code>if</code>	Start a conditionally executed block of statements
<code>select</code>	<code>switch</code>	Start a multi-branch block of statements
<code>while</code>	<code>while</code>	Start repeated execution of a block while a condition is satisfied

But there is more than the semantic difference between keywords `switch` and `otherwise` in Matlab and `select` and `else`, respectively, in Scilab. The following comparison illustrates this difference. With function `foobar` defined as:

```
function foobar(a)
// Scilab
select a
case ['foo','pipo']
    disp('ok')
case 'foo'
    disp('not ok')
else
    disp('invalid case')
end
endfunction
```

we get

```
-->foobar(['foo','pipo'])
ok

-->foobar('foo')
not ok

-->foobar('pipo')
invalid case
```

The variable `a` following the keyword `select` can be any Scilab data object.

The analogous Matlab function

```
function foobar(a)
% Matlab
switch a
case {'foo', 'pipo'}
    disp('ok')
case 'foo'
    disp('not ok')
otherwise
    disp('invalid case')
end
```

on the other hand, leads to

```
>>foobar({'foo', 'pipo'})
??? SWITCH expression must result in a scalar or string constant.

>>foobar('foo')
ok

>>foobar('pipo')
ok
```

The variable `a` following the keyword `switch` can only be a scalar or string constant. On the other hand, a `case` can represent more than one value of the variable. The strings `'foo'` and `'pipo'` satisfy the first case and so the second case is never reached.

In an `if` clause Scilab has the optional keywords `then` and `do` as in

```
-->if a >= 0 then a=sqrt(a); end

-->if a >= 0 do a=sqrt(a); end
```

but `then` and `do` can be replaced by a comma, a semicolon, or a `<RETURN>`. Hence, both statements are equivalent to

```
-->if a >= 0, a=sqrt(a); end
```

Likewise, the `for` loop can be written with the optional keyword `do` as in

```
for i = 1:n do a(i)=asin(2*%pi*i); end
```

and again `do` can be replaced by a comma, a semicolon, or a `<RETURN>`. The same is true for the `while` clause.

Matlab uses the `try/catch` syntax to trap errors. Its functionality can be emulated by means the combination of `errcatch` and `iserror`. This is illustrated in the following code fragment<sup>1</sup>. For the sake of clarity it is shown here the way it would look in a file.

```

errcatch(-1,'continue','nomessage'); // Start error trapping 3
a=1/0 4a
if iserror() // Check for error
    disp('A division by zero has occurred')
    errclear(-1)
end

a=1/0 4b
b=1
errclear(-1)
errcatch(-1) // Error trapping toggled off
a=1/0 4c

```

Statement `3` starts error trapping with the system error message suppressed. Statements `4a`, `4b`, and `4c` represent errors. Execution of these statements produces the following result:

```

-->errcatch(-1,'continue','nomessage'); // Start error trapping 3

-->a=1/0 4a

-->if iserror() // Check for error
--> disp('A division by zero has occurred')

A division by zero has occurred
--> errclear(-1)
-->end

-->

-->a=1/0 4b

-->b=1
b =
    1.

```

<sup>1</sup>`errcatch` is considered “fragile”, and thus this construct should be used only for debugging; for more robust code the use of `execstr(...,'errcatch')` is preferable.

```

-->errclear(-1)

-->errcatch(-1)    // Error trapping toggled off

-->a=1/0                                                    4c
    |--error    27
division by zero...

```

Clearly, the three identical “division by zero” errors are treated differently. The first one, [4a](#), is trapped and the user-supplied message is printed; the second, [4b](#), is trapped and ignored; the third division by zero, [4c](#), occurs after error trapping has been turned off and creates the standard system error message.

Other commands can be considered as at least related to flow control. They include **pause** which interrupts execution similar to Scilab’s **keyboard** command, but with some important differences explained beginning on page 15.

Another function, **halt**, can be used to temporarily interrupt a program or script. Execution of a function or script will stop upon encountering **halt()** and wait for a key press before continuing.

### 3.5 General Functions

Table 3.3 lists Scilab’s low-level functions and commands (commands are actually functions used with command-style syntax; see Section 5.1). Some, like **date**, are essentially the same as in Matlab, others have slightly different names (**exists** vs. **exist**), some may have the same name but may give slightly different output (in Scilab **length** with a matrix argument returns the product of the number of rows and columns, in Matlab **length** returns the larger of the number of rows and columns), and many are quite different.

In this list of functions the command **pause** deserves special attention. It is equivalent to Matlab’s **keyboard** command in that it interrupts the flow of a script or function and returns control to the keyboard. However, a Matlab function/script stays in the workspace of the function. In Scilab the **pause** command creates a new workspace. The prompt changes from, say, **-->** to **-1->** where the number 1 indicates the workspace level. All variables of all lower workspace are available at this new workspace as long as they are not shadowed (a variable in a lower workspace is shadowed if a variable with the same name is defined in a higher workspace). This is an example:

```

-->a = 1, b = 2    // Variables in original workspace
a =
  1.
b =
  2.

```

```

-->pause // Creates new workspace (level 1)

-1->disp([a,b])
! 1. 2. !

-1->c = a+b
c =
3.

-1->a = 0
a =
0.

-1->return // Return to original workspace 5a

-->a, c
a =
1.
!--error 4
undefined variable : c

```

Scilab	Description
\$	Index of last element of matrix or (row/column) vector
apropos	Keyword search for a function
clear	Clear unprotected variables and functions from memory
clearglobal	Clear global variables from memory
date	Current date as string
disp	Display input argument
getdate	Get date and time in an 8-element numeric vector
global	Define variables as global
halt	Stop execution and wait for a key press
help	On-line help
inttype(a)	Output numeric code representing type of integer <b>a</b>
pause	Interrupt execution of function or script
timer	Outputs time elapsed since the preceding call to <b>timer()</b>
type(a)	Output numeric code representing type of variable <b>a</b>
typeof(a)	Output string with type of variable <b>a</b>
whereis	Display name of library containing function
who	Displays/outputs names of current variables
whos	Displays/outputs names and specifics of current variables

Table 3.3: General functions



The command `pause` creates a new workspace (the level of this workspace becomes part of the prompt); the display function `disp` shows that the variables `a` and `b` are available in this new workspace, and the new variable `c` is computed correctly. However, upon returning to the original workspace we find that `a` still has the value 1 (in spite of being changed in the level-1 workspace) and that the variable `c` is not available anymore. This is not what we would have found with Matlab's `keyboard` command.

In order to get these new values to the original workspace they have to be passed on by the `return` command. In Scilab the `return` command can have input and output arguments!

```
-->a = 1, b = 2
a =
1.
b =
2.

-->pause // Create new workspace (level 1)

-1->disp([a,b])
! 1. 2. !

-1->c = a+b
c =
3.

-1->a = 0
a =
0.

-1->[aa,c] = return(a,c) // Return to original workspace 5b

-->aa,c
aa =
0.
c =
3.
```

The above two code fragments are identical except for the `return` statements `5a` and `5b`. Statement `5b` returns variables `a` and `c` created in the level-1 workspace renaming `a` to `aa`. Without this change, the existing variable `a` would have overwritten by the value of `a` created in the level-1 workspace. A more complicated use of the `return` function is illustrated in statement `29` on page 82.

The command `resume` is equivalent to the `return` command (we could have used `resume` instead of `return` in the two examples above).

Unlike its Matlab counterpart, the display function `disp`, which has already been used above, allows more than one input parameter:

```
-->disp(123,'The result is:')  
The result is:  
123.
```

It shows the same behavior noted previously: the input arguments are printed to the screen beginning with the last.

## Chapter 4

# Variable Types

The only two variable types a casual user is likely to define and use are numeric variables and strings; but Scilab has many more data types — in fact, it has more than Matlab. Hence, it is important to be able to identify them. Unlike Matlab, which uses specific functions with boolean output for each variable type (e. g. *iscell*, *ischar*, *isnumeric*, *issparse*, *isstruct*), Scilab uses essentially two functions, **type** and **typeof**: the former has numeric output the other outputs a character string. The following table lists variable types and the output of **type** and **typeof**. The last column of this table, with heading “Op-type”, defines the abbreviation used to specify the operand type for operator overloading (see Section 7.3).

Type of variable	<b>type</b>	<b>typeof</b>	Op-type
real or complex constant matrix	1	'constant'	s
polynomial matrix	2	'polynomial'	p
boolean matrix	4	'boolean'	b
sparse matrix	5	'sparse'	sp
sparse boolean matrix	6	'boolean sparse'	spb
Matlab sparse matrix	7	?	msh
matrix of integers stored in 1, 2, or 4 bytes	8	Depends on type of integer	i
matrix of character strings	10	'string'	c
function (un-compiled)	11	'function'	m
function (compiled)	13	'function'	mc
function library	14	'library'	f
list	15	'list'	l
typed list ( <b>tlist</b> )	16	Depends on type of list	tl
matrix-like typed list ( <b>mlist</b> )	17	Depends on type of list	ml
pointer	128	?	ptr
index vector with implicit size	129	'size implicit'	ip

Table 4.1: Variable types

In addition, there is a special function (see Table 4.2) for variables of type integer (see Table 4.3). For variables of type integer the function `typeof` outputs a character string identical to the name of the function that creates it (see Table 4.3). Thus

```
-->i8=uint8(16)      // i8 is an unsigned 8-bit integer
i8 =
    16

-->typeof(i8)
ans =
uint8
```

The output of `typeof` for typed lists (`tlist`) and matrix-like typed lists (`mlist`) is discussed in Section 4.5.

Function `typeof` affords a straightforward simulation of Matlab function `isa`:

```
>> i8=uint8(11);
>> isa(i8,'uint8')
ans =
    1
```

and

```
-->i8=uint8(11);
-->typeof(i8) == 'uint8'
ans =
    T
```

are equivalent.

Scilab	Description
<code>inttype(a)</code>	Output numeric code representing type of integer <code>a</code>
<code>type(a)</code>	Output numeric code representing type of variable <code>a</code>
<code>typename</code>	Associate a variable type name with a numeric avariable type
<code>typeof(a)</code>	Output string with type of variable <code>a</code>
<code>who</code>	Displays/outputs names of current variables
<code>whos</code>	Displays/outputs names and specifics of current variables

Table 4.2: Utility functions for managing/classifying of variables

## 4.1 Numeric Variables — Scalars and Matrices

Scilab knows matrices. This term includes scalars and vectors. Scalars and the elements of vectors and matrices can be real or complex. The statements

```
-->a = 1.2;
-->b = 1.0e3;
-->cx = a+%i*b
    cx  =
      1.2 + 1000.i
```

define three  $1 \times 1$  matrices, i.e. scalars. There is no function like `complex` in Scilab. Defining a complex number, such as `cx` above, is done via an arithmetic statement.

Vectors and matrices can be entered and accessed in much the same way as in Matlab.

```
-->mat=[ 1 2 3; 4 5 6]
    mat  =
    !  1.   2.   3. !
    !  4.   5.   6. !

-->mat2=[mat;mat+6]
    mat2  =
    !  1.   2.   3. !
    !  4.   5.   6. !
    !  7.   8.   9. !
    ! 10.  11.  12. !

-->mat(2,3)
    ans  =
      6.

-->mat(2,:)
    ans  =
    !  4.   5.   6. !

-->mat($,$)
    ans  =
      6.

-->mat($)
    ans  =      6.
```

There is a difference in the way the last element of a vector or matrix is accessed. Scilab uses the `$` sign as indicator of the last element whereas Matlab uses `end`. The `$` represents, in fact, a somewhat more powerful concept and can be used to create an implied-size vector, a new variable of type `'size implicit'`.

```
-->index=2:$
index =
2:1:$

-->type(index)
ans =
129.

-->typeof(index)
ans =
size implicit

-->mat2(2,index)
ans =
! 5. 6. !
```

There is no equivalent in Matlab for this kind of addressing of matrix elements.

By default, Scilab variables are double-precision floating point numbers. However, like Matlab, Scilab also knows integers. Conversion functions are shown in Table 4.3. Function `iconvert`, which takes two input arguments, does essentially what the seven other conversion functions listed in this table do.

Scilab	Description
<code>double</code>	Convert integer array of any type/length to floating point array
<code>iconvert</code>	Convert numeric array to any integer or floating point format
<code>int8(a)</code>	Convert <code>a</code> to an 8-bit signed integer
<code>int16(a)</code>	Convert <code>a</code> to a 16-bit signed integer
<code>int32(a)</code>	Convert <code>a</code> to a 32-bit signed integer
<code>uint8(a)</code>	Convert <code>a</code> to an 8-bit unsigned integer
<code>uint16(a)</code>	Convert <code>a</code> to a 16-bit unsigned integer
<code>uint32(a)</code>	Convert <code>a</code> to a 32-bit unsigned integer

Table 4.3: Integer conversion functions

Matlab allows no mathematical operations for such integers. Scilab is more lenient and lets the user worry about wrap-around and possibly other problems.

```

-->u = int8(100), v = int8(2)
u =
100
v =
2

-->u*v
ans =
-56

```

The result is wrapped (200-256). Unsigned integers give an analogous result.

```

-->x = uint8(100), y = uint8(2), z= uint8(3)
idxuint8
x =
100
y =
2
z =
3

-->x*y
ans =
200

-->x*z
ans = 44

```

Again, the last result is wrapped (300-256). Operations between integers of different type are not allowed, but those with standard (double precision) floating point numbers are, and the result is a floating point number.

```

-->typeof(z)
ans =
uint8

-->typeof(2*z)
ans =
constant

```

The variable `z`, defined in the previous example, is an unsigned one-byte integer. Multiply it by 2 and the result is a regular floating point number for which `typeof` returns the value `constant`.

## 4.2 Special Matrices

Like Matlab, Scilab has a number of functions that create “standard” matrices. Many of them have the same or a very similar names. The arguments or the output may be slightly different.

The empty matrix `[]` in Scilab behaves slightly different than the corresponding `[]` in Matlab; in Scilab, for example,

```
-->a = []+3 6a
a =
  3.
```

whereas in Matlab

```
>>a = []+3 6b
a =
 [].
```

Scilab	Description
<code>[]</code>	Empty matrix
<code>companion</code>	Companion matrix
<code>diag</code>	Create diagonal matrix or extract diagonal from matrix
<code>eye</code>	Identity matrix (or its generalization)
<code>grand</code>	Create random numbers drawn from various distributions
<code>ones</code>	Matrix of ones
<code>rand</code>	Matrix of random numbers with uniform or normal distribution
<code>sylv</code>	Sylvester matrix (input two polynomials, output numeric)
<code>testmatrix</code>	Test matrices: magic square, Franck matrix, inverse of Hilbert matrix
<code>toeplitz</code>	Toeplitz matrix
<code>zeros</code>	Matrix of zeros

Table 4.4: Special matrices

While 6a is the default result, Scilab’s behavior in this situation can be changed by invoking Matlab-mode.

```
-->mtlb_mode(%t)
-->a = []+3 6c
a =
 []
```



With Matlab-mode true, Scilab `6c` produces the same result Matlab `6b` does.

The standard syntax with two arguments to define dimensions works for functions `zeros`, `ones`, `rand`, `eye` the same way it does for Matlab.

```
-->rand_mat = rand(2,3)
rand_mat =
!   .2113249   .0002211   .6653811 !
!   .7560439   .3303271   .6283918 !
```

However, the syntax used in the following example

```
-->a=[1 2 3; 4 5 6];

-->rand_mat = rand(a) 7
rand_mat =
!   .2113249   .0002211   .6653811 !
!   .7560439   .3303271   .6283918 !
```

has been deprecated in Matlab; it expects `7` written as `rand_mat = rand(size(a))`.

## 4.3 Character String Variables

### 4.3.1 Creation and Manipulation of Character Strings

Character strings can be defined with single quotes or double quotes, but the opening quote must match the closing quote. Thus `8a` and `8b` below are equivalent.

```
-->test = "This is a test"; 8a
```

```
-->test = 'This is a test'; 8b
```

Function `length` produces a familiar result—the number of characters in the string.

```
-->length(test)
ans =
    14.
```

However, a character string in Scilab is not a vector but rather akin to a Matlab cell. Thus

```
-->size(test)
ans =
!   1.   1. !
```

This is the same result `size` would give in Matlab if `test` were a Matlab cell. Thus it is not surprising that strings can be elements of matrices.

```
-->sm = ['This is','a','matrix';
-->      'each element','is a','string']
sm =
!This is      a      matrix !
!                                     !
!each element is a string !

-->size(sm)
ans = !  2.  3. !
```

Not surprisingly, function `size` again gives the same result `size` would give for a Matlab cell array. In other words: in order to get an analogous representation in Matlab one would have to use a cell array. However, there is no analog in Matlab for the behavior of `length`; nevertheless, it is a straight-forward generalization of its behavior for an individual string.

Scilab	Description
<code>ascii</code>	Convert ASCII codes to equivalent string and vice versa
<code>convstr</code>	Convert string to lower or upper case
<code>date</code>	Current date as string
<code>emptystr</code>	Create a matrix of empty strings
<code>grep</code>	Find matches of strings in a vector of strings
<code>gsort(a)</code>	Sort elements/rows/columns of <code>a</code>
<code>intersect(str1,str2)</code>	Returns elements common to two vectors <code>str1</code> and <code>str2</code>
<code>length</code>	Matrix of lengths of the strings in a string matrix
<code>msprintf</code>	Convert, format, and write data to a string
<code>msscanf</code>	Read variables from a string under format control
<code>part</code>	Extract substrings from a string or string vector
<code>pol2str</code>	Convert polynomial to a string
<code>sci2expr</code>	Convert a variable into a string representing an expression
<code>size</code>	Size/dimensions of a Scilab object
<code>strcat</code>	Concatenate character strings
<code>strindex(str1,str2)</code>	Return starting index/indices where string <code>str2</code> occurs in <code>str1</code>
<code>string</code>	Convert numbers to strings
<code>stripblanks</code>	Remove leading and trailing blanks from a string
<code>strsubst(s1,s2,s3)</code>	Substitute string <code>s3</code> for <code>s2</code> in <code>s1</code>
<code>union(a,b)</code>	Extract the unique common elements of <code>a</code> and <code>b</code>
<code>unique(a)</code>	Return the unique elements of <code>a</code> in ascending order

Table 4.5: Functions that manipulate strings

```

-->length(sm)
ans =
!   7.   1.   6. !
!  12.   4.   6. !

```

The output of `length` is a matrix with the same dimension as `sm`; each matrix entry is the number of characters of the corresponding entry of `sm`. For many purposes this output is so useful that one misses it in Matlab.

A handy function for many string manipulations is `ascii` which converts character strings into their ASCII equivalent (e.g. 'A' to 65, 'a' to 97) and vice versa. In fact, beginning with Scilab 2.6, it even supports the 8-bit ASCII standard (including the Euro which recently replaced the “international currency symbol”). Function `ascii` does in Scilab what the pair `char` and `double` does in Matlab.

Below is an example where `ascii` is used to emulate Matlab function `isletter`.

```

function bool=isletter(str)
// Function creates a boolean vector bool the length of which is
// equal to the number of characters in string str.
// An element of bool is true if the corresponding character in
// str is a letter and false if it is not not.
// INPUT
// str    character string
// OUTPUT
// bool   boolean vector

    temp = ascii(str);
    bool = (temp >= 65 & temp <= 90) | (temp >= 97 & temp <= 122);
endfunction

```

With this function

```

-->isletter('abc123 END.')
ans =
! T T T F F F F T T T F !

```

Strings can be concatenated by means of the `+` sign

```

-->s1 = 'Example';

-->s2 = 'of ';

-->s3 = 'concatenation';

```

```
-->ss1 = s1+' '+s2+s3'      9a
ss1 =
Example of concatenation    10a
```

```
-->length(ss1)
ans =      24.
```

The following is also a legal Scilab statement; it creates a string matrix.

```
-->ss2 = [s1,' ',s2,s3]      9b
ss2 =
!Example of concatenation !  10b

-->length(ss2)
ans =
!  7.   1.   3.   13. !
```

In Scilab, statement [9a](#) does what [9b](#) would do in Matlab; in Scilab the variable `ss2` is a 4-element string vector, and Scilab's built-in function `strcat` can be used to convert string vector `ss2` to string `ss1`. Note the difference in the display of `ss1` [10a](#) and `ss2` [10b](#). The exclamation marks in [10b](#) indicate that `ss2` is a string vector. In Matlab, strings in cell arrays are in quotes.

The operator `+` works for string matrices the same way it works for numeric matrices. As illustrated below, a single string “added” to a string matrix is prepended (or appended) to every element.

```
-->cost = ['10' '100' '1000'; '1' '13' '-22']
cost =
!10 100 1000 !
!           !
!1  13  -22  !

-->new_cost= '$'+cost+'.00'
new_cost =
!$10.00 $100.00 $1000.00 !
!           !
!$1.00  $13.00  $-22.00  !
```

With this kind of indexing the question is how one would access individual characters in a string. As far as extracting characters is concerned this can be done with function `part`.

```
-->test = 'This is a test';
-->part(test,1) // Select the first character
```

```

ans =
T

-->part(test,1:4) // Select the first 4 characters
ans =
This

-->str1 = part(test,11:20) 11a
str1 =
test

-->length(str1)
ans =
10.

```

The second argument of `part` is a vector of indices. For every index that exceeds the length of the string a blank is appended to the output of `part`. This is illustrated in 11a; `str1` consists of the requested 10 characters, and 11b below shows that the last six characters of `str1` are indeed blanks (ASCII code 32).

```

-->ascii(str1) 11b
ans =
! 116. 101. 115. 116. 32. 32. 32. 32. 32. 32. !

```

This property provides one way to emulate Matlab's function `blanks` and create a string of blanks by extending the empty string

```

-->blanks = part(emptystr(),1:10) // Create a string of 10 blanks
blanks =

-->ascii(blanks)
ans =
! 32. 32. 32. 32. 32. 32. 32. 32. 32. 32. !

```

Scilab's `stripblanks` command is not quite like Matlab's `deblank` in that the latter only removes trailing blanks. A quick and dirty, but vectorized, implementation of `deblank` — at least for single strings — would be

```

function strout = deblank(str)
// Function removes trailing blanks from input string
index = find(ascii(str) ~= 32);
strout = part(str,1:index($));
endfunction

```

which again uses functions `part` and `ascii`. This example again illustrates the use of `$` to denote the last element of a vector — analogous to the use of `end` in Matlab as the last index. The simple modification required in the last statement to remove leading blanks (or leading and trailing blanks) is obvious.

It is worth reviewing a few more of the functions shown in Table 4.5.

`emptystr()` returns an empty string or string matrix very similar to the function `cell` in Matlab.

```
-->length(emptystr(2,5))
ans =
!  0.  0.  0.  0.  0.  !
!  0.  0.  0.  0.  0.  !
```

`grep(vstr1,str2)` searches for occurrence of string `str2` in string vector `vstr1`; returns a vector of indices of those elements of `vstr1` where `str2` has been found or an empty vector if `str2` does not exist in any element of `vstr1`.

`strindex(str1,vstr2)` looks for the position in string `str1` of the character string(s) in string vector `vstr2`. Function `strindex` differs from `grep` in that its first input argument is a string whereas the first argument of `grep` can be a string vector. The index vector output by `grep` refers to elements of the string vector `vstr1` whereas the index vector output by `strindex` refers to the position of the elements of `vstr2` in string `str1`. This is illustrated by the following example.

```
-->str1 = 'abcdefghijkl';

-->idx1 = grep(str1,'jk')           //String 'jk' is in str1(1)
idx1 =
    1.

-->idx2 = strindex(str1,'jk')      //String 'jk' is in str at position 10
idx2 =
    10.

-->str2 = ['abcdefghijkl','xyz','jklm'];

-->idx3 = grep(str2,'jk')         //String 'jk' is in str2(1) and str2(3)
idx3 =
!   1.   3. !

-->idx4 = grep(str2,['jk','y'])   //Strings 'jk' or 'y' are in str2(1)
                                   //str2(2), and str2(3)
idx4 =
!   1.   2.   3. !
```

This means that `grep` with some additional checking can be used to emulate the Matlab function `ismember` for string arguments (see page 39).

Like its Matlab equivalent the function `msscanf` can be use to extract substrings separated by spaces and numbers from a string.

```
-->str=' Weight: 2.5 kg';
-->[a,b,c] = msscanf(str,'%s%f%s')
c =
kg
b =
2.5
a =
Weight:

-->typeof(a)
ans =
string

-->typeof(b)
ans =
constant
```

The format types available are `%s` for strings, `%e`, `%f`, `%g` for floating-point numbers, `%d`, `%i` for decimal integers, `%u` for unsigned integers, `%o` for octal numbers, `%x` for hexadecimal numbers, and `%c` for a characters (white spaces are not skipped). For more details see the help file for `scanf_conversion`.

In the context of the next subsection the function `sci2exp` may come in handy. It converts a variable into a string representing a Scilab expression. An example is

```
-->a=[1 2 3 4]'
a =
! 1. !
! 2. !
! 3. !
! 4. !

-->stringa = sci2exp(a)
stringa =
[1;2;3;4]
```

### 4.3.2 Strings with Scilab Expressions

Like Matlab, Scilab allows execution of strings with Scilab statements and expressions. There are three possible functions with slightly different features

Scilab	Matlab	
<code>eval</code>	<code>eval</code>	Evaluate string vector with Scilab expressions
<code>evstr</code>	<code>eval</code>	Evaluate string vector with Scilab expressions
<code>execstr</code>	<code>eval</code>	Evaluate string vector with Scilab expressions or statements

While there is a Scilab function `eval`, the best functional equivalent to Matlab's `eval` is `execstr`. Thus

```
-->execstr('a=1+sin(1)')

-->a
a =
    1.841471
```

Note that the `execstr` does not echo the result even though there is no semicolon at the end of the statement. A more elaborate use of `execstr` is

```
-->ier = execstr(['a=2', 'b=3^a'], 'errcatch', 'n')
ier =
    0.

-->a,b
a =
    2.
b =
    9.
```

This code fragment illustrates that the first input argument of `execerr` can be a string vector. Since the second input argument `'errcatch'` is given, an error in one of the statements of the first argument does not issue an error message. Instead, `execstr` aborts execution at the point where the error occurred, and resumes with `ier` equal to the error number. In this case the display of the error message is controlled by the third input argument (`'m'` ==> error message, `'n'` ==> no error message).

An example of this use of `execstr` with the `errcatch` option is the simulation of a search path for the execution of a Scilab script on page 81.

In Scilab `eval` evaluates a vector of Scilab expressions. Thus

```
-->c = eval(['1+sin(1)'; '1+cos(1)'])
```



```

c =
!  1.841471 !
!  1.5403023 !

```

Note that `eval('a=1+sin(1)')` produces the error message

```

Warning: obsolete use of = instead of ==
%h = a=1+sin(1)
!
at line      2 of function %eval          called by :
line   16 of function eval              called by :
eval('a=1+sin(1)')
!--error    4
undefined variable : b
at line      2 of function %eval          called by :
line   17 of function eval              called by :
eval('a=1+sin(1)')

```

Scilab expects an expression and interprets the `=` as a typo, assumes that the user really means `==`, and then finds that `b` is undefined.

The Scilab command `evstr` is very similar to `eval`; it, too, only works with expressions. However, while `eval` has no provisions to allow user-defined error handling, `evstr` will trap errors if used with two output arguments.

```

-->[c,ier] = evstr(['1+sin(1)';'1+cos(1)']) 12b
ier =
    0.
c =
!  1.841471 !
!  1.5403023 !

```

If an error occurs, `ier` is set to the error number, but the function does not abort. The following is an example where the second expression of the of the argument has a syntax error.

```

-->[c,ier] = evstr(['1+sin(1)';'1+-cos(1)'])
ier =
    2.
c =
[]

```

The function does not abort, but `ier` is set to 2.

Note: since all the variables of the whole workspace (that are not shadowed) are available to these three functions there is generally no need for an equivalent to Matlab function `evalin`.

### 4.3.3 Symbolic String Manipulation

Scilab has several function that treat strings as variables and perform symbolic operations. An examples is `trianfml` which converts a symbolic matrix to upper triangular form.

```
-->mat = ['a', 'b+c', 'd'; '-b*a', '0', 'a+b'; 'b', '1', '-1']
mat =
!a      b+c  d      !
!              !
!-b*a  0      a+b  !
!              !
!b      1      -1   !

-->tri = trianfml(mat)
tri =
!b  1      -1              !
!              !
!0  b*a  b*(a+b)-b*a      !
!              !
!0  0      b*a*(b*d+a)-(b*(b+c)-a)*(b*(a+b)-b*a) !
```

A symbolic matrix can be evaluated by means of the function `evstr` discussed above.

```
-->a = 1, b = -1, c = 2, d = 0
a =
  1.
b =
 - 1.
c =
  2.
d =
  0.

-->nummat = evstr(tri)
nummat =
! - 1.    1.   - 1.  !
!  0.   - 1.    1.  !
!  0.    0.    1.  !
```

There are several functions such as `solve` and `trisolve` that operate on symbolic matrices and `addf`, `subf`, `mulf`, `ldivf`, and `rdivf` that operate on symbols representing scalars. What exactly they do can be found by looking at their help files.

## 4.4 Boolean Variables

Boolean variables are represented by `%t` (true) and `%f` (false). Since Scilab's main initialization file equates the corresponding upper-case and lower-case variables they can also be used with capital letters (`%T`, `%F`). This is different from Matlab where boolean variables, while intrinsically different from arithmetic numbers, are represented by 1 and 0. This analogy is illustrated in the following table which shows a line-by-line comparison of corresponding Scilab and Matlab statements.

Scilab	Matlab
<code>--&gt;n = [1 1]</code> <code>n =</code> <code>! 1. 1. !</code>	<code>&gt;&gt; n = [1 1]</code> <code>n =</code> <code>1 1</code>
<code>--&gt;b = [%t,%t]</code> <code>b =</code> <code>! T T !</code>	<code>&gt;&gt; b = logical([1,1])</code> <code>b =</code> <code>1 1</code>
<code>--&gt;a = [1 2]</code> <code>a =</code> <code>! 1. 2. !</code>	<code>&gt;&gt; a = [1 2]</code> <code>a =</code> <code>1 2</code>
<code>--&gt;a(n)</code> <code>ans =</code> <code>! 1. 1. !</code>	<code>&gt;&gt; a(n)</code> <code>ans =</code> <code>1 1</code>
<code>--&gt;a(b)</code> <code>ans =</code> <code>! 1. 2. !</code>	<code>&gt;&gt; a(b)</code> <code>ans =</code> <code>1 2</code>

When displayed on the screen in Matlab, vectors `n` and `b` look exactly the same. Nevertheless, they are different

```
>> islogical(n)
0
>> islogical(b)
1
```

and, when used as indices for the vector `a`, they produce different results. But, fortunately, these results agree with those obtained with Scilab.

There are three operators, well known from Matlab, that operate on boolean variables.

&	Logical AND
~	Logical NOT
	Logical OR

In the proper context, both Scilab and Matlab treat numeric variables like logical variables; any real numeric variable  $\neq 0$  is interpreted as **true** and 0 is interpreted as **false**. Thus

```
-->if -1.34
--> a=1;
-->else
--> a=2;
-->end
-->a
a =
    1.
```

Interestingly, Scilab itself is not very consistent regarding the use of boolean variables. The two functions **exists** and **isdef** do the same thing: they check if a variable exists. However, **isdef** outputs **T** if the variable exists and **F** otherwise, whereas **exists** outputs 1 and 0, respectively. In this sense the function **bool2s** can be considered as having boolean output. If **a** is a numeric matrix, then **b = bool2s(a)** creates a matrix **b** where all non-zero entries of **a** are 1. If **a** is boolean then **b** is 1 where **a** is **%t** and 0 where **a** is **%f**. The same result — even without an execution-time penalty — can be achieved by adding 0 to the boolean matrix **a**.

Since there is no Scilab function analogous to **zeros** or **ones** a similar trick must be used to create a boolean matrix or vector.

```
-->>false = ~ones(1,10)
false =
! F F F F F F F F F F !

-->>true = ~~ones(1,10)
true =
! T T T T T T T T T T !
```

One could, in principle, define **true** as **true=~zeros(1,10)**, i. e. without the double negation, but in Scilab-2.6 function **zeros** is much slower than functions **ones**. This is expected to change in Scilab-2.7.

In Matlab all this would work as well but **logical(ones(n,m))** is faster.

Like in Matlab, boolean variables can be used in arithmetic expressions where they act as if they were 1 and 0, respectively.

```
-->5*%t
```

```

ans =
    5.

-->3*%f
ans =
    0.

```

Table 4.6 lists a number of functions that output or use boolean variables.

Functions `and` and `or` are functional equivalents of Matlab's functions `all` and `any`, respectively.<sup>1</sup> Function `and(a)` returns the boolean variable `%t` if all entries of `a` are `%t` (for a boolean matrix) or non-zero (for a numeric matrix).

```

a =
!  0.   1. !
!  2.   3. !

-->and(a)
ans =
    F

-->and(a, 'r') 13
ans =
! F T !

```

In the example above `a` has a zero entry in the upper left corner; hence, the answer is `false`. With the optional second argument `'r'` (line 13), `and` analyzes the columns of `a` and outputs a row vector. The first column contains a zero; hence the first element of the output vector is `f`. Matlab's `all` would output an analogous logical vector.

Function `or(a)` returns the boolean variable `%t` if at least one entry of `a` is `%t` (for a boolean matrix) or non-zero (for a numeric matrix). Hence, for the same matrix `a`

```

-->or(a)
ans =
    T

-->or(a, 'c')
ans =
! T !
! T !

```

<sup>1</sup>See also the discussion of `max`, `min`, etc. on page 56

Scilab	Description
<code>MSDOS</code>	Variable is <code>%t</code> if computer is PC and <code>%f</code> otherwise
<code>and(a)</code>	Output <code>%t</code> if all entries of the boolean matrix <code>a</code> are true
<code>bool2s</code>	Replace <code>%t</code> (or non-zero entry) in matrix by 1 and <code>%f</code> by zero
<code>exists(a)</code>	Test if variable <code>a</code> exists
<code>find(a)</code>	Find the indices for which boolean matrix <code>a</code> is true
<code>isdef(a)</code>	Test if variable <code>a</code> exists
<code>iserror</code>	Test if error has occurred
<code>isglobal(a)</code>	Test if <code>a</code> is a global variable
<code>isinf(a)</code>	Test if <code>a</code> is infinite
<code>isnan(a)</code>	Output boolean vector with entries <code>%t</code> where <code>a</code> is <code>%nan</code>
<code>isreal(a)</code>	Test if <code>a</code> is real (“or if its imaginary part is “small”)
<code>mtlb_mode</code>	Test for (or set) Matlab mode for empty matrices
<code>or(a)</code>	Output <code>%t</code> if at least one entry of the boolean matrix <code>a</code> is true
<code>simp_mode</code>	Test for (or set) simplification mode for rational expressions

Table 4.6: Functions that operate on, or output, boolean variables

With the second argument `'c'` function `or` analyzes the rows and puts out a column vector. Since each row has at least one non-zero element, all entries of the output are `%t`. The analog to Matlab’s `any` is `or` with the second input argument `'r'`.

Like Matlab, Scilab always evaluates all terms of a logical expression; it does not, say, evaluate an expression from left to right and stop as soon as the result is no longer in question. Thus the statement

```
bool = exists('a') & a > 0
```

will fail with an error message if the variable `a` does not exist even though the fact that `exists('a')` is false also means that `bool` is false. This statement needs to be split up.

```
bool = exists('a')
if bool
    bool = a > 0;
end
```

Some of the constructs discussed above are used in the following example of an emulation — but only for string vectors/matrices — of the Matlab function `ismember`. For example, the Matlab statements

```
>>vstr = {'abcd', 'abc', 'xyz', 'uvw'};
>>str = 'abc';
>>index = ismember(vstr, str)
index =
    0    1    0    0
```

produce the same result as the analogous Scilab statements

```
-->vstr = ['abcd','abc','xyz','uvw'];
-->str = ['abc','xy'];
-->index = ismember(vstr,str)
index =
! F T F F !.
```

if the function `ismember` is defined as

```
function bool=ismember(strv1,strv2)
// Function outputs a boolean vector the same size as strv1.
// bool(i) is set to %t if the string strv1(i) is equal to
// one of the strings in strv2

bool = ~ones(strv1); // Create a boolean vector %f
[idx1,idx2]=grep(strv1,strv2); // Find indices of strv1 and strv2
// for which there is a match

if idx1 == []
return
end

// Eliminate indices for which an element of strv2 is only
// a substring in strv1

temp1 = strv1(idx1);
temp2 = strv2(idx2);
bool(idx1(temp1(:) == temp2(:))) = %t;
endfunction
```

## 4.5 Lists

Lists are Scilab data objects and come in three flavors: ordinary lists, `list`, which behave like Matlab cell vectors (one-dimensional cell arrays), typed lists, `tlist`, and matrix-oriented typed lists, `mlist`. The latter two can be used to emulate Matlab structures.

### 4.5.1 Ordinary lists (`list`)

A list is a collection of data objects. Its Matlab equivalent is a vector of cells. Like Matlab cell arrays these objects need not be of the same type. They can be scalars, matrices, character strings, string matrices, functions, as well as other lists. An example is (remember that both single quotes (') and double quotes (") can be used to denote strings in Scilab):

```

-->a_list=list('Test',[1 2; 3 4], ...
               ['This is an example'; 'of a list entry'])
a_list =

    a_list(1)
Test

    a_list(2)
!  1.  2.  !
!  3.  4.  !

    a_list(3)
!This is an example !
!                   !
!of a list entry    !

```

Individual elements can be accessed with the usual index notation. Thus

```

-->a_list(1)
ans =
Test

```

This is different from the way Matlab works. If `a_list` were a Matlab cell array the same result would be achieved by `a_list{1}` — note the curly brackets — whereas `a_list(1)` would be a one-element cell array which contains the string `'test'`.

Using the index 0 one can prepend an element to the list

```

-->a_list(0)=%eps;

```

Scilab	Description
<code>getfield</code>	Get a data object from a list
<code>length</code>	Length of list
<code>list</code>	Create a list
<code>lstcat</code>	Concatenate lists
<code>mlist</code>	Create a matrix-oriented typed list
<code>null</code>	Delete an element of a list
<code>setfield</code>	Set a data object in a list
<code>size</code>	Size of a list or typed list (but not matrix-oriented typed list)
<code>tlist</code>	Create a typed list

Table 4.7: Functions that create, or operate on, lists



This pushes all elements of `a_list` back. Hence

```
-->a_list(2)
ans =
Test
```

What used to be the first element is now the second one. The Matlab equivalent would be `a_list=[{eps},a_list]`; it is more flexible since any number of elements (not just one) could be prepended and the augmented cell array could be saved under a new name; e.g.

```
a_list1=[{eps},a_list]
```

However, in Scilab the same functionality could be created by overloading (see Section 7.3).

Appending elements works the same way.

```
-->a_list(8)='final element';
```

assigns the string `'final element'` to element 8 of the list `a_list`. Elements 5 to 7 are undefined. Thus

```
-->a_list(5)
!--error 117 List element 5 is Undefined
```

The `null` function can be used to delete an elements of a list. For example,

```
-->aa=list(1,2,3,4,5);

-->aa(3)=null()
aa =
aa(1)
1.
aa(2)
2.
aa(3)
4.
aa(4)
5.
```

The third element has been removed from the list. The list has now only four elements. It is not possible to delete more than one element at a time in this way; e.g. the attempt to delete elements 2 and 4 via `aa([2,4])=null()` generates an error message.

Lists allow tuple assignments, i. e. more than one variable can be assigned a value in a single statement. With the list `aa` defined above

```

-->[u,v]=aa(2:3)
v =
4.
u =
2.

```

This kind of tuple assignment can also be used with typed lists.

The functions `size` and `length` have been appropriately overloaded for lists.

```

-->blist = list('abcd','efg',1.3,[1 2; 3 4],list('1',1))
blist =
      blist(1)
abcd

      blist(2)
efg

      blist(3)
1.3

      blist(4)
!  1.  2.  !
!  3.  4.  !

      blist(5)

      blist(5)(1)
1
      blist(5)(2)
1.

-->length(blist)
ans =
5.

-->size(blist)
ans =
5.

```

Note that `length` and `size` give the same result — one number. Lists are inherently one-dimensional objects. But this last example illustrates how one can emulate a two-dimensional cell array, i. e. a multi-dimensional object where an element is defined by two indices (this may be

desirable for tables where some columns have alphanumeric entries while others are purely numeric). One can write it as a list of lists. The following is an example.

```
-->cell=list(list(),list());

-->cell(1)=['first','second','third'];

-->cell(2)=[1,2,3];

-->cell(1)(3)
ans =
third

-->cell(2)(3)
ans =
3.
```

Nevertheless,

```
-->length(cell)
ans =
2.
```

Thus `cell` is still a one-dimensional data object.

#### 4.5.2 Typed lists (tlist)

A typed list is a special kind of list. Typed lists allow the user to set up special kinds of data objects and to define operations on these objects (see Section 7.3). Examples are linear systems (type `'lls'`) or rational functions (type `'rational'`; see page 53).

The first element of a typed list must be a string (the type name or type) or a vector of strings. In the latter case the type name is the first element of the string vector; the other elements of this string vector are names (field names) for the other entries of the typed list. An example is

```
-->a_tlist=tlist(['example','first','second'], 1.23,[1,2])
a_tlist =

    a_tlist(1)
!example first second !

    a_tlist(2)
1.23
```

```

        a_tlist(3)
!   1.   2. !

-->type(a_tlist)
ans =
16.

-->typeof(a_tlist)
ans =
example

```

Here the first element of the list is a three-element vector of character strings whose first element, `'example'`, identifies the type of list (type name). While this type name can consist of almost any number of characters (definitely more than 1024), it must not have more than 8 if one intends to overload operators for this typed list.

From a Matlab user's perspective the fact that typed lists can be used to represent Matlab structures is of greatest relevance here, and in this case the type as represented by the first element of the first string vector can, in principle, be ignored<sup>2</sup>. The other elements of the first string vector play the role of the field names of the structure. The elements of `a_tlist` can be accessed in the usual way via indices.

```

-->a_tlist(1)
ans =
!example first second !

-->a_tlist(2)
ans =
1.23

-->a_tlist(3)
ans =
!   1.   2. !

-->a_tlist(1)(2)
ans =
first

```

Displays of lists can become quite lengthy and confusing. Here, for display purposes a function `show` is used (it is not part of the Scilab distribution, but too long to be reproduced here) which

---

<sup>2</sup>As shown above, the display of lists can be rather unwieldy. Fortunately, the way a typed list (or matrix-oriented typed list) is displayed can be overloaded to create, for example, a Matlab-like look. If this is desired then the type name plays a key role.

displays data objects in a more compact form and, for typed lists, is patterned after the format Matlab uses for structures. Thus

```
-->show(a_tlist)
LIST OF TYPE "example"
  first: 1.23
  second: 1  2
```

Section 7.3 shows how this kind of display can be made the default for displaying a particular type of a typed list.

Elements of the typed list other than the first can be accessed in various ways. For example

```
-->a_tlist('first')
ans =
    1.23

-->a_tlist('second') 14a
ans =
!  1.  2. !
```

Thus the second and third element of `a_tlist(1)` can be used as “names” for the second and third element, respectively, of `a_tlist`. But there is another way of using these names. It is the representation of structures familiar to Matlab and C users.

```
-->a_tlist.first
ans =
    1.23

-->a_tlist.second 14b
ans =
!  1.  2. !
```

Thus a typed list can be accessed like a Matlab structure . Once it is defined, different values can be assigned to it in the same way they would be assigned to a Matlab structure.

```
-->a_tlist.second = 'A new value';

-->a_tlist.second
ans =
A new value
```

One advantage of 14a over 14b is that the field name need not satisfy requirements for a variable; it may contain white spaces and special characters not allowed for variable names. But more

importantly, the field name may be computed by concatenating strings or it could be the element of a string vector.

In principle, the typed list `a_tlist` could have been defined as

```
-->a_tlist = tlist(['example','first','second']);

-->a_tlist.first = 1.23;

-->a_tlist.second = [1,2];
```

In contrast to Matlab, where the fields of a structure need not be defined before they are used, in Scilab one must define them. If `a_tlist` were to have one more element, it would have to be added first — e.g via (remember that `$` means “last element” equivalent to `end` in Matlab);

```
-->a_tlist(1)($+1) = 'new';

-->a_tlist.new = 'value of new field'; 15a;

-->show(a_tlist)
LIST OF TYPE "example"
  first: 1.23
  second: 1 2
  new: value of new field
```

The statement 15a above could have been written as

```
a_tlist($+1) = 'value of new field'; 15b
```

Generally speaking, the  $k$ th element of the first-element character string vector of a typed list is the field name of the  $k$ th element of the typed list.

Lists can have other lists as elements. For example

```
-->record=tlist(['record','patient','invoice']);

-->record.patient=tlist(['patient','address','city','phone']);

-->record.patient.phone='123.456.7890';

-->record.invoice=1234.33;

-->record
record =
```

```

        record(1)
!record patient invoice !

        record(2)

        record(2)(1)
!patient address city phone !

        record(2)(2)
Undefined

        record(2)(3)
Undefined

        record(2)(4)
123.456.7890

        record(3)
1234.33

```

With function `show` this reads:

```

-->show(record)
LIST OF TYPE "record"
  patient: LIST OF TYPE "patient"
    address: Undefined
    city: Undefined
    phone: 123.456.7890
  invoice: 1234.33

```

An element of a typed list can be removed the same way an element of an ordinary list is removed. However, the index or the name can be used. Thus, for the typed list `record` defined above the following four Scilab statements

```

-->record.patient.phone = null();
-->record.patient(4) = null();
-->record(2)(4) = null();
-->record(2).phone = null();

```

are equivalent.

A combination of `list` and `tlist` can be used to create the Scilab equivalent of a structure array.

```

-->seis1 =
tlist(['seismic','first','last','step','traces','units'],0,[],4,[],'ms');

-->seismic = list(seis1,seis1,seis1);

-->for ii=1:3
--> seismic(ii).last=1000*ii;
--> nsamp = (seismic(ii).last-seismic(ii).first)/seismic(ii).step+1;
--> seismic(ii).traces=rand(nsamp,10);
-->end

-->show(seismic)
List element 1:
LIST OF TYPE "seismic"
  first: 0
  last: 1000
  step: 4
  traces: 251 by 10 matrix
  units: ms
List element 2:
LIST OF TYPE "seismic"
  first: 0
  last: 2000
  step: 4
  traces: 501 by 10 matrix
  units: ms
List element 3:
LIST OF TYPE "seismic"
  first: 0
  last: 3000
  step: 4
  traces: 751 by 10 matrix
  units: ms

```

Thus `seismic` is a list with three seismic data sets with the same start times but different end times, that can be individually addressed.

```

-->show(seismic(3))
LIST OF TYPE "seismic"
  first: 0
  last: 3000
  step: 4
  traces: 751 by 10 matrix

```



```
units: ms
```

It is also straight forward to access fields of individual data sets. For example,

```
-->seismic(2).last
ans =
    2000.
```

### 4.5.3 Matrix-oriented typed lists (mlist)

Help file and manuals provide only very sketchy information about matrix-oriented typed lists. An mlist is defined like a regular typed list discussed above. This is illustrated by an example. The statement

```
-->an_mlist=mlist(['VVV','name','value'],['a','b','c'],[1 2 3])
an_mlist =

      an_mlist(1)
!VVV name value !

      an_mlist(2) 16a
!a b c !

      an_mlist(3)
! 1. 2. 3. !
```

creates a matrix-like typed list, and the statements

```
-->an_mlist.name
ans = !a b c !

-->an_mlist('name')
ans =
!a b c !

-->an_mlist.value
ans =
! 1. 2. 3. !

-->an_mlist('value')
ans =
! 1. 2. 3. !
```

work as expected. However, elements cannot be accessed by index the way elements of a typed list can.

```
-->an_mlist(2) 16b
                !--error      4
                undefined variable : %1_e
```

This is in spite of the fact that `16b` looks exactly like `16a`, the output created by function `mlist`. Also, the `size` function does not work with mlists. In practical terms, this implies more options for operator overloading. And, indeed, matrix-oriented typed lists appear to be better suited for operator overloading.

## 4.6 Polynomials

If polynomials are a data type available with standard Matlab (there is, of course, the Symbolic Toolbox based on the Maple kernel) then, at least, I am not aware of them. In Scilab they can be created by means of function `poly`.

```
-->p = poly([1 2 3], 'z', 'coeff')
p =
      2
    1 + 2z + 3z

-->typeof(z)
ans =
    polynomial

-->typeof(p)
ans =
    polynomial
```

In this example the first argument of `poly` is a vector of polynomial coefficients. Alternatively, it is also possible to define a polynomial via its roots.

Scilab	Description
<code>bezout</code>	Compute greatest common divisor of two polynomials
<code>clean</code>	Round to zero small entries of a polynomial matrix
<code>cmdred</code>	Create common-denominator form of two polynomial matrices
<code>coeff</code>	Compute coefficients of a polynomial matrix
<code>coffg</code>	Compute inverse of a polynomial matrix
<code>colcompr</code>	Column compression of polynomial matrix
<code>degree</code>	Compute degree of polynomial matrix
<code>denom</code>	Compute denominator of a rational matrix
<code>derivat</code>	Compute derivative of the elements of a polynomial matrix
<code>det</code>	Compute determinant of a polynomial or rational matrix
<code>determ</code>	Compute determinant of a polynomial matrix
<code>detr</code>	Compute determinant of a polynomial or rational matrix
<code>diophant</code>	Solve diophantine equation
<code>factors</code>	Compute factors of a polynomial
<code>gcd</code>	Compute greatest common divisor of elements of polynomial matrix
<code>hermit</code>	Convert polynomial matrix to triangular form
<code>horner</code>	Evaluate polynomial or rational matrix
<code>hrmt</code>	Compute greatest common divisor of polynomial row vector
<code>inv</code>	Invert rational or polynomial matrix
<code>invr</code>	Invert rational or polynomial matrix
<code>lcm</code>	Compute least common multiple elements of polynomial matrix
<code>lcmdiag</code>	Least common multiple diagonal factorization
<code>ldiv</code>	Polynomial matrix long division
<code>pdiv</code>	Elementwise polynomial division of one matrix by another
<code>pol2str</code>	Convert polynomial to a string
<code>residu</code>	Compute residues (e. g. for contour integration) of ratio of two polynomials
<code>roots</code>	Compute roots of a polynomial
<code>rowcompr</code>	Row compression of polynomial matrix
<code>sfact</code>	Spectral factorization of polynomial matrix
<code>simp</code>	Rational simplification of elements of rational polynomial matrix
<code>simp_mode</code>	Test for (or set) simplification mode for rational expressions
<code>sylm</code>	Sylvester matrix (input two polynomials, output numeric)

Table 4.8: Functions related to polynomials and rational functions

```

-->p = poly([1 2 3], 'z', 'roots')
p =
          2   3
    - 6 + 11z - 6z + z

-->roots(p)
ans =
!   1. !
!   2. !
!   3. !

```

The default for the third argument is actually `'roots'` and so it could have been omitted. It is also possible to define first the symbolic variable and then create polynomials via standard Scilab expressions.

```

-->s = poly(0, 's') // This is a polynomial whose only zero is 0
s =
s

-->p = 2 - 3*s + s^2
p =
          2
    2 - 3s + s

-->q = 1 - s
q =
1 - s

-->simp_mode(%f) // Do not simplify ratios of polynomials

-->r = p/q
r =
          2
    2 - 3s + s
-----
    1 - s

-->simp_mode(%t) // Simplify ratios of polynomials

-->simp(r)
ans =
    2 - s
-----

```

```
1  
  
-->type(r)  
ans =  
16.  
  
-->typeof(r)  
ans =  
rational
```

The result of `type` indicates that `r` is a typed list and `typeof` tells us that it is a list of type `rational`.

Table 4.8 lists functions available in Scilab for manipulating polynomials and ratios of polynomials.

One difference between computer algebra packages such as Mathematica, Maple, or Macsyma and this implementation of polynomial algebra is the precision. Scilab evaluates expression to its normal precision while the above packages maintain infinite precision unless requested to perform numerical evaluations.

# Chapter 5

## Functions

### 5.1 General

For someone coming from Matlab Scilab functions are familiar entities. One difference is that parentheses are generally required even if a function has no input arguments. There are two exceptions:

- The function is treated as a variable
- The function has at most one output argument and all input arguments are strings (command-type syntax).

**Command-style Syntax:** For any function that has at most one output argument and whose input arguments are character strings, the calling syntax may be simplified by dropping the parentheses. Thus

```
-->getf('fun1.sci')
-->getf 'fun1.sci'
-->getf fun1.sci // Command-style syntax
```

are equivalent. The last form represents the command-style syntax (a command, possibly followed by one or more arguments; Matlab has a similar feature). More generally, if function `funct` accepts three string arguments then

```
funct('a','total','of three strings')
```

is equivalent to

```
funct a total 'of three strings'
```

Here the quotes around the last argument are required to prevent it from being interpreted as three individual strings. It even seems to work if the function accepts non-string arguments provided that these arguments are optional. In order to run a script, say `script.sce`, the command `exec('script.sce')` must be executed. The function `exec` has one required and two optional arguments (one of which is numeric). Nevertheless,

```
exec('script.sce')
exec 'script.sce'
exec script.sce
```

give all the same result.

Scilab provides one way of passing parameters to a function that is not available in Matlab: named arguments. This method of passing arguments is especially practical with function that have many input parameters with good default values — plot functions are typical examples. For example, the built-in function `plot2d` can be called as follows

```
plot2d([logflag],x,y,[style,strf,leg,rect,nax])
```

The first argument is an optional string that can be used to set axis graduation (linear or logarithmic). The next two arguments are the x-coordinate and y-coordinate of the function to be plotted. The last five arguments are optional again. Now suppose one wants to use the default values for all optional parameters except the curve legend (parameter `leg`). The parameter `logflag` is not a problem. If the first input argument is not a string the program knows it is not given as a positional parameter. But the defaults of `style` and `strf` would have to be given so that `leg` is at the correct position in the argument list. Hence, the statement would read as follows

```
-->plot2d(x,y,1,'161','Curve legend')
```

This, of course means that one has to figure out what the default values are. The simpler solution to this problem is to use named parameters

```
-->plot2d(x,y,leg='Curve legend')
```

Note that the name of the argument, `leg` is not quoted — it is not a string. The order of named parameters is arbitrary, but any positional parameters must come before named parameters. It is for example possible to specify the parameter `logflag` after all. For example,

```
-->plot2d(x,y,leg='Curve legend',logflag='ll')
```

creates the same plot, but with log-log axes. Of course, the same could be achieved by

```
-->plot2d('ll',x,y,leg='Curve legend')
```

In principle, any input argument could be supplied as a named parameter.

```
-->plot2d(x=x,y=y,leg='Curve legend')
```

but `plot2d` has internal checks that do not allow that. Also, named parameters are not compatible with variable-length input argument lists `varargin`.

## 5.2 Functions that Operate on Scalars and Matrices

### 5.2.1 Basic Functions

Quite a number of functions in Table 5.1 below, while having the same name, behave differently for matrices than their Matlab counterparts. The following example illustrates this difference. Scilab has an edge here.

```
-->mat = matrix([1:20],4,5) // Create a matrix by rearranging a vector
mat =
!  1.   5.   9.  13.  17. !
!  2.   6.  10.  14.  18. !
!  3.   7.  11.  15.  19. !
!  4.   8.  12.  16.  20. !

-->[maxa,index] = max(mat) // Find largest element and its location
index =
!  4.   5. !
maxa =
    20.

-->[maxr,idx] = max(mat,'r')
idx =
!  4.   4.   4.   4.   4. !
maxr =
!  4.   8.  12.  16.  20. !

-->maxc = max(mat,'c')
maxc =
!  17. !
!  18. !
!  19. !
!  20. !
```



Scilab	Description
<code>abs(a)</code>	Absolute value of <b>a</b> , $ a $
<code>bool2s</code>	Replace <b>%t</b> (or non-zero entry) in matrix by 1 and <b>%f</b> by zero
<code>ceil(a)</code>	Round the elements of <b>a</b> to the nearest integers $\geq a$
<code>clean</code>	“Clean” matrices; i.e. small entries are set to zero
<code>conj</code>	Complex conjugate
<code>cumprod</code>	Cumulative product of all elements of a vector or array
<code>cumsum</code>	Cumulative sum of all elements of a vector or array
<code>fix(a)</code>	Rounds the elements of <b>a</b> to the nearest integers towards zero
<code>floor(a)</code>	Rounds the elements of <b>a</b> to the nearest integers $\geq a$
<code>gsort(a)</code>	Sort elements/rows/columns of <b>a</b>
<code>imag</code>	Imaginary part of a matrix
<code>intersect(str1,str2)</code>	Returns elements common to two vectors <b>str1</b> and <b>str2</b>
<code>lex_sort</code>	Sort rows of matrices in lexicographic order
<code>linspace</code>	Create vector with linearly spaced elements
<code>logspace</code>	Create vector with logarithmically spaced elements
<code>max</code>	Maximum of all elements of a vector or array
<code>maxi</code>	Maximum of all elements of a vector or array
<code>mean</code>	Mean of all elements of a vector or array
<code>median</code>	Median of all elements of a vector or array
<code>min</code>	Minimum of all elements of a vector or array
<code>mini</code>	Minimum of all elements of a vector or array
<code>modulo(a,b)</code>	$a-b.*\text{fix}(a./b)$ if $b \sim 0$ ; remainder of <b>a</b> divided by <b>b</b>
<code>pmodulo(a,b)</code>	$a-b.*\text{floor}(a./b)$ if $b \sim 0$ ; remainder of <b>a</b> divided by <b>b</b>
<code>prod</code>	Product of the elements of a matrix
<code>real</code>	Real part of a matrix
<code>round(a)</code>	Round the elements of <b>a</b> to the nearest integers
<code>sign(a)</code>	Signum function, $a/ a $ for $a \neq 0$
<code>sqrt(a)</code>	$\sqrt{a}$
<code>st_deviation</code>	Standard deviation
<code>sum</code>	Sum of all elements of a matrix
<code>union(a,b)</code>	Extract the unique common elements of <b>a</b> and <b>b</b>
<code>unique(a)</code>	Return the unique elements of <b>a</b> in ascending order

Table 5.1: Basic arithmetic functions

There is no equivalent in Matlab for the behavior of `max(mat)`. It is particular the easy way of getting the indices of the largest element of a matrix that I consider extremely useful. The Matlab version of `max` behaves just like `max(mat,'r')` does — it computes a row vector representing the maxima of every column. Likewise, `max(mat,'c')` computes a column vector with the maximum element in each row (equivalent to `max(mat,[],2)` in Matlab).

The analogous behavior is found for Scilab functions `cumprod`, `cumsum`, `maxi`, `mean`, `median`, `min`, `mini`, `prod`, `sum`, and `st_deviation`. The functions `max` and `maxi` are equivalent as are `min` and `mini`.

Scilab has two functions for sorting: `gsort` and `sort`. The latter is not only buggy (BOT, Chapter 6.1.3.3.6) but also less powerful. Hence, it is only `gsort` that is discussed here. Unlike its Matlab counterpart, `gsort` sorts in decreasing order by default. It also behaves differently for matrices. While Matlab sorts each column, Scilab sorts all elements and then stores them columnwise as shown in the example below.

```
-->mat = [-1 4 -2 2;1 0 -3 3]
mat =
! - 1.    4.   - 2.    2. !
!  1.    0.   - 3.    3. !

-->smat = gsort(mat)
smat =
!  4.    2.    0.   - 2. !
!  3.    1.   - 1.   - 3. !

-->smatc = gsort(mat,'c') // Rows are sorted 17
smatc =
!  4.  2.  - 1.  - 2. !
!  3.  1.  0.  - 3. !
```

In the help file 17 is called a “columnwise” sort; this appears to be somewhat misleading since — as described later in the help file — the rows are the ones that are being sorted. The first column contains the largest element of each row, the second column the second largest, etc. Thus  $\text{smatc}(:,i) \geq \text{smatc}(:,j)$  for  $i < j$ .

A third input parameter allows the user to select the sort direction (decreasing or increasing) of `gsort`. In order to get what Matlab’s `sort` would do one needs to set it to increasing (`'i'`) and also choose “row sorting” (`'r'`).

```
-->smatri = gsort(mat,'r','i') // Matlab-like result
smatri =
! - 1.    0.   - 3.    2. !
!  1.    4.   - 2.    3. !
```

This way the elements of each column are sorted in increasing order.

Function `gsort` also has an option to perform a lexicographically increasing or decreasing sort. This corresponds to Matlab’s `sortrows` command and is illustrated below for sorting of rows

```
-->mat1 = [3 4 1 4; 1 2 3 4; 3 3 2 1; 3 3 1 2]
mat1 =
!  3.    4.    1.    4. !
!  1.    2.    3.    4. !
```

```

!   3.   3.   2.   1. !
!   3.   3.   1.   2. !

//      Lexicographically increasing sorting of rows
-->[smat1,index] = gsort(mat1,'lr','i')
      index =
!   2. !
!   4. !
!   3. !
!   1. !
      smat1 =
!   1.   2.   3.   4. !
!   3.   3.   1.   2. !
!   3.   3.   2.   1. !
!   3.   4.   1.   4. !

```

The first column is sorted first. Rows that have the same element in the first column are sorted by the entries of the second column. If two or more of those are the same as well the entries of the third column are used to determine the order, etc. The optional second output argument gives the sort order (thus `smat1 = mat1(index,:)`).

Changing input argument `'lr'` to `'lc'` changes row sorting to column sorting.

While shown here for numeric arrays, string arrays can be sorted the same way.

### 5.2.2 Elementary Mathematical Functions

Except for `cotg` the names of all the elementary transcendental functions listed in Table 5.2 agree with those of their Matlab counterparts. Furthermore, `atan` can be called with one or with two arguments. With one argument it equivalent to Matlab's `atan`; with two arguments it corresponds to Matlab's `atan2`. Thus, for  $x > 0$ , `atan(y,x) == atan(y/x)`.

If the argument of any of these functions is a matrix, the function is applied to each entry separately. Thus

```

-->a = [1 2; 3 4]
      a =
!   1.   2. !
!   3.   4. !
-->b = sqrt(a)
      b =
!   1.           1.4142136 !
!   1.7320508   2.           !

```

```

-->b.*b
ans =
!  1.  2. !
!  3.  4. !

```

18a

Scilab	Description
<code>acos</code>	Arc cosine
<code>acosh</code>	Inverse hyperbolic cosine
<code>asin</code>	Arc sine
<code>asinh</code>	Inverse hyperbolic sine
<code>atan</code>	Arc tangent
<code>atanh</code>	Inverse hyperbolic tangent
<code>cos</code>	Cosine
<code>cosh</code>	Hyperbolic cosine
<code>cotg</code>	Cotangent
<code>coth</code>	Hyperbolic cotangent
<code>exp</code>	Exponential function
<code>log</code>	Natural logarithm
<code>log10</code>	Base-10 logarithm
<code>log2</code>	Base-2 logarithm
<code>sin</code>	Sine
<code>sinh</code>	Hyperbolic sine
<code>tan</code>	Tangent
<code>tanh</code>	Hyperbolic tangent

Table 5.2: Elementary transcendental functions

The functions listed in Table 5.3 are “true” matrix functions; they operate on a matrix as a whole. Thus the matrices have to satisfy certain requirements, the minimum being that they must be square. So, the example above, with same matrix `a` but for `sqrtn`, looks like this

```

-->b = sqrtn(a)
b =
!  .5536886 + .4643942i  .8069607 - .2124265i !
!  1.2104411 - .3186397i  1.7641297 + .1457544i !

```

```

-->b*b
ans =
!  1. + 5.551E-17i  2. !
!  3. + 2.776E-17i  4. !

```

18b

```

clean(b*b)      19
ans =
!  1.  2.  !
!  3.  4.  !

```

Obviously, the matrix **b** is complex and so rounding errors lead to small imaginary parts of some of the entries in the product **b\*b**. Expression [19](#) illustrates how function **clean** can be used to remove such small matrix entries.

The important difference between these two examples is that in [18a](#) corresponding entries of **b** are multiplied (the **.** in front of the **\***) whereas in [18b](#) the matrices are multiplied.

The list of functions in Table 5.3 is longer than it would be in Matlab; on the other hand Scilab lacks an equivalent for Matlab's **funm** function which works for any user-specified functions; for good accuracy, matrices should be symmetric or Hermitian.

Scilab	Description
<b>acoshm</b>	Matrix inverse hyperbolic cosine
<b>acosm</b>	Matrix arc cosine
<b>asinhm</b>	Matrix inverse hyperbolic sine
<b>atanhm</b>	Matrix inverse hyperbolic tangent
<b>atanhm</b>	Matrix inverse hyperbolic tangent
<b>coshm</b>	Matrix hyperbolic cosine
<b>cosm</b>	Matrix cosine
<b>expm</b>	Matrix xponential function
<b>logm</b>	Matrix natural logarithm
<b>sinhm</b>	Matrix hyperbolic sine
<b>sinm</b>	Matrix sine
<b>sqrtnm</b>	Matrix square root
<b>tanhm</b>	Matrix hyperbolic tangent
<b>tanm</b>	Matrix tangent

Table 5.3: Matrix functions

### 5.2.3 Special Functions

Table 5.4 lists so-called special functions of mathematical physics available in Scilab.

Scilab	Description
<code>%asn</code>	Jacobian elliptic function, $\text{sn}(x, m) = \int_0^x dt / \sqrt{(1-t^2)(1-mt^2)}$
<code>%k</code>	Complete elliptic integral, $K(m) = \int_0^1 dt / \sqrt{(1-t^2)(1-mt^2)}$
<code>%sn</code>	Jacobian elliptic function, $\text{sn}$
<code>amell</code>	Jacobian function $\text{am}(u, k)$
<code>besseli</code>	Modified Bessel function of the first kind, $I_\alpha(x)$
<code>besselj</code>	Bessel function of the first kind, $J_\alpha(x)$
<code>besselk</code>	Modified Bessel function of the second kind, $K_\alpha(x)$
<code>bessely</code>	Bessel function of the second kind, $Y_\alpha(x)$
<code>delip</code>	Elliptic integral, $u(x, k) = \int_0^x dt / \sqrt{(1-t^2)(1-k^2)}$
<code>dlgamma</code>	Digamma function, $\psi(x) = d \ln(\Gamma(x)) / dx$
<code>erf</code>	Error function, $\text{erf}(x) = 2/\sqrt{\pi} \int_0^x \exp(-t^2) dt$
<code>erfc</code>	Complementary error function, $\text{erfc}(x) = 2/\sqrt{\pi} \int_x^\infty \exp(-t^2) dt$
<code>erfcx</code>	Scaled complementary error function, $\text{erfcx}(x) = \exp(x^2) \text{erfc}(x)$
<code>gamma</code>	Gamma function, $\Gamma(x) = \int_0^\infty t^{x-1} \exp(-t) dt$
<code>gammaln</code>	Logarithm of the Gamma function, $\ln(\Gamma(x))$

Table 5.4: Special functions

Scilab	Description
<code>bandwr</code>	Band-width reduction of a sparse matrix
<code>chfact</code>	Sparse Cholesky factorization
<code>chsolve</code>	Use sparse Cholesky factorization to solve linear system of equations
<code>full</code>	Convert sparse to full matrix
<code>lufact</code>	Sparse LU factorization
<code>luget</code>	Sparse LU factorization
<code>lusolve</code>	Solve sparse linear system of equations
<code>nnz</code>	Number of nonzero elements of a sparse matrix
<code>sparse</code>	Create sparse matrix
<code>spchol</code>	Sparse Cholesky factorization
<code>speye</code>	Sparse identity matrix
<code>spget</code>	Retrieve entries of a sparse matrix
<code>spones</code>	Replace non-zero elements in sparse matrix by ones
<code>sprand</code>	Create sparse random matrix
<code>spzeros</code>	Sparse zero matrix

Table 5.5: Functions for sparse matrices

### 5.2.4 Linear Algebra

Tables 5.5 and 5.6 list functions for linear-algebra operations. Functions for full matrices work on sparse matrices as well.

Scilab	Description
<code>balanc</code>	Balance matrix to improve condition number
<code>bdiag</code>	Block diagonalization of matrix
<code>bdiag(M)</code>	Block diagonalization/generalized eigenvectors of <b>M</b>
<code>chol(M)</code>	Choleski factorization; $R' * R = M$
<code>colcomp(M)</code>	Column compression of <b>M</b>
<code>cond</code>	Condition number of <b>M</b>
<code>det</code>	Determinant of a matrix
<code>fullrf(M)</code>	Full-rank factorization of <b>M</b>
<code>fullrfk(M)</code>	Full-rank factorization of $M^K$
<code>givens</code>	Given's rotation
<code>hess(M)</code>	Hessenberg form of <b>M</b>
<code>householder</code>	Householder orthogonal reflection matrix
<code>inv(M)</code>	Inverse of matrix <b>M</b>
<code>kernel(M)</code>	Nullspace of <b>M</b>
<code>linsolve</code>	Linear-equation solver
<code>norm(M)</code>	Norm of <b>M</b> (matrix or vector)
<code>orth(M)</code>	Orthogonal basis for the range of <b>M</b>
<code>pinv(M)</code>	Pseudoinverse of <b>M</b>
<code>polar(M)</code>	Polar form of <b>M</b> , $M=R*\expm(\%i*Theta)$
<code>qr(M)</code>	QR decomposition of <b>M</b>
<code>range(M)</code>	Range of <b>M</b>
<code>rank(M)</code>	Rank of <b>M</b>
<code>rcond(M)</code>	Reciprocal of the condition number of <b>M</b> ; L-1 norm
<code>schur(M)</code>	Schur decomposition
<code>spaninter(M,N)</code>	Intersection of the span of <b>M</b> and <b>N</b>
<code>spanplus(M,N)</code>	Span of <b>M</b> and <b>N</b>
<code>spec</code>	Eigenvalues of matrix
<code>sva(M)</code>	Singular-value approximation of <b>M</b> for specified rank
<code>svd(M)</code>	Singular-value decomposition of <b>M</b>
<code>trace(M)</code>	Trace (sum of diagonal elements) of <b>M</b>

Table 5.6: Linear algebra

### 5.2.5 Signal-Processing Functions

Scilab proper and the Signal-Processing Toolbox offer quite a number of functions for signal processing. The three functions shown here in table 5.7 have been chosen because they are frequently used and have Matlab-equivalents. Furthermore, the Fast Fourier Transform (FFT) `fft` may need some explanation. After all, it does what `fft`, `ifft`, `fft2`, and `ifft2` do in Matlab.

Scilab	Description
<code>convol</code>	Convolution
<code>fft</code>	Forward and inverse Fast Fourier Transform
<code>mfft</code>	Multidimensional Fast Fourier Transform

Table 5.7: Functions for signal processing

The basic Fourier transform is performed as shown in the example below

```
-->x = rand(100,1);

-->y = fft(x,-1);      20a

-->z = fft(y,1);      20b

-->norm(x-z)
ans =
1.532E-15
```

where

$$y_m = \sum_{n=1}^N x_n e^{-2\pi i(n-1)(m-1)/N} \quad \text{for } m = 1, \dots, N \quad (5.1)$$

with  $N$  denoting the number of elements  $x_n$ . The second argument, -1, in `20a` corresponds to the minus sign in front of the exponent in (5.1). The operation performed in `20b`,

$$z_m = \frac{1}{N} \sum_{n=1}^N y_n e^{2\pi i(n-1)(m-1)/N} \quad \text{for } m = 1, \dots, N,$$

is the inverse of `20a`.

If `xx` is a matrix then `f(xx,-1)` performs the two-dimensional Fourier transform. It is thus equivalent to Matlab's `fft2`. Matlab's `fft`, on the other hand, performs a one-dimensional FFT on each column of a matrix. In order to achieve the same result with Scilab one has to write `fft` in the form `21`



```

-->n = 100; m = 20;

-->xx = rand(n,m);

-->yy1 = zeros(xx);

-->for i=1:m
-->  yy1(:,i) = fft(xx(:,i),-1);
-->end

-->yy2 = fft(xx,-1,n,1);      [21]

-->norm(yy1-yy2)              [22]
ans =
0.

-->zz = fft(yy2,1,n,1);      [23]

-->norm(xx-zz)                [24]
ans =
3.368E-15

```

Expression [22] shows that `yy1` and `yy2` are identical. Likewise, expression [24] shows that the inverse Fourier transform [23] works as expected with this syntax.

Furthermore, with `yy2` computed in [21], statement [25] computes the two-dimensional FFT of `xx`:

```

-->uu1 = fft(xx,-1); // Two-dimensional FFT of xx

-->uu2 = fft(yy2,-1,m,n);    [25]

-->norm(uu1-uu2)
ans =
0.

```

Obviously, `uu1` and `uu2` are identically.

## 5.3 File Input and Output

There are quite a few functions for formatted and unformatted reading and writing of text and numeric data. Some have Matlab equivalents. They are summarized in tables 5.9 (reading), 5.10 (writing), and 5.8 (ancillary functions). Many I/O functions come in pairs — one is designed to read what the other one writes. The special-purpose routines for, say, writing and reading audio files fall into this category. Some of the following pairs represent my own way of grouping. This grouping does not imply that no other function can read what one of these functions writes and vice versa; rather, these pairs appear similar in terms of design philosophy and input arguments.

### 5.3.1 Opening and Closing of Files

Before one can read from, or write to, a file the file needs to be “opened”. This is transparent for I/O functions, such as `fprintfMat` or `fscanfMat`, that only use a file name to specify which file to read from (write to). They open the requested file, read/write the data and close the file without the user being aware of it. But whenever there is a need to incrementally read or write data it is up to the user to open files. A situation like that occurs, for example, with big data sets. One might read a piece of the data from file A, process it, and write it out to file B; then read in the next piece of data from file A, process it, and write it to file B, etc. In this case files A and B must be opened before anything can be read from respectively written to them. Scilab has two functions for opening a file, `mopen` and `file`, and Scilab functions that allow incremental I/O require one or the other. For this reason the subsequent discussion of specific I/O functions mentions, where appropriate, which one of the two functions needs to be used for opening a file. Function `mopen` is quite similar to Matlab’s `fopen` whereas `file` reminds one of the Fortran equivalent.

Functions `mopen` and `file` output a file identifier (Matlab terminology). Scilab help files call it “file descriptor” or “logical unit descriptor”; in Fortran it is called “Logical Unit Number”. It is this file identifier, and not the file name, that is then used to specify from which file to read (to which file to write). File identifiers are numbers which range from 1 to 19 in Scilab-2.6 for Windows. File identifier 1 is used for the history file `scilab.hist`, file identifiers 5 and 6 (`%io(1)` and `%io(2)`, respectively) are reserved for keyboard (input) and Scilab window (output), respectively. Hence, a maximum of 16 file identifiers are available to users; this limits to 16 the number of user files that can be open at any one time.

Files that have been opened with `mopen` must be closed with `mclose`, and `file` with the `'close'` option must be used to close files that have been opened with `file`. Examples of the use of `mopen`, `mclose`, and `file` are part of the discussion of specific I/O functions below.

Scilab	Description
<code>dispfiles</code>	Display properties of opened files
<code>file</code>	Open/close a file, define file attributes
<code>fileinfo</code>	Get information about a file
<code>getio</code>	Get Scilab's standard logical input/output units
<code>mclearerr</code>	Reset binary-file access errors
<code>mclose</code>	Close (all) open file(s)
<code>meof</code>	Check if end-of-file has been reached
<code>mopen</code>	Open a file
<code>mseek</code>	Set position in a binary file
<code>mtell</code>	Output the current position in a binary file
<code>newest</code>	Find newest of a set of files
<code>xgetfile</code>	Open dialog box for file selection

Table 5.8: Functions that open, query, manipulate, and close files

### 5.3.2 Functions `mgetl` and `mputl`

Function `mputl` writes a vector of strings to an ASCII file in form of a sequence of lines, and `mgetl` can retrieve one or more of these lines. This is a simple example:

```
-->text = ['This is line 1';'Line 2 ','Line 3 (last)']
text =
!This is line 1 !
!                !
!Line 2          !
!                !
!Line 3 (last)  !
-->mputl(text,'C:\temp\dummy.txt')
-->
-->all = mgetl('C:\temp\dummy.txt') // Get the whole file
all =
!This is line 1 !
!                !
!Line 2          !
!                !
!Line 3 (last)  !
```

With only one input argument, `mgetl` reads the whole file. If only the first few lines are required the number of lines can be specified via the second input parameter:

Scilab	Description
<code>auread</code>	Read a .au audio file from disk
<code>excel2sci</code>	Read ASCII file created by MS Excel
<code>fscanf</code>	Read numeric/string variables from ASCII file under format control
<code>fscanfMat</code>	Read matrix from ASCII file
<code>input</code>	Read from keyboard with a prompt message to Scilab window
<code>load</code>	Load variables previously saved with <code>save</code>
<code>loadwave</code>	Read a .wav sound file
<code>mfscanf</code>	Read data from file (C-type format)
<code>mget</code>	Read numeric data (vector) from binary file (conversion format)
<code>mgeti</code>	Read data from binary file, converts to Scilab integer format
<code>mgetl</code>	Read a specified number of lines from ASCII file
<code>mgetstr</code>	Read bytes from binary or ASCII file and interpret as character string
<code>mscanf</code>	Read data from keyboard (C-type format)
<code>mtlb_load</code>	Load variables from file with Matlab-4 format
<code>read</code>	Read matrix of strings/numbers from ASCII file under format control
<code>read4b</code>	Read Fortran binary file (4 bytes/word)
<code>readb</code>	Read Fortran binary file (8 byte/word)
<code>readc_</code>	Read a character string from a file/keyboard
<code>save</code>	Save current Scilab variables in binary file
<code>wavread</code>	Read a .wav sound file

Table 5.9: Functions that input data from files or keyboard

```
-->only2 = mgetl('C:\temp\dummy.txt',2) // Read first 2 lines only
only2 =
!This is line 1 !
!           !
!Line 2      !
```

If more lines are requested than are available, the function aborts with an error message. If the second argument is -1, all lines are read (equivalent to no second input argument).

In the examples above the file to use is identified by its name. In a case like this `mgetl` does three things. It opens the file for reading, reads the lines requested, and closes the file again. This convenience comes at a price. It is not possible to read a file a few lines at a time. If this is necessary you must open the file yourself and use the file identifier created by `mopen` to specify the file to `mgetl`. Finally, once you are done reading, you need to close the file again.

```
fid = mopen('C:\temp\dummy.txt','r') // Open file for reading
fid =
3.
-->one = mgetl(fid,1) // Read one line
one =
```

Scilab	Description
<code>auwrite</code>	Write a .au audio file to disk
<code>diary</code>	Write screen output of a Scilab session to a file
<code>disp</code>	Write input argument to Scilab window
<code>fprintf</code>	Write formatted data to file (like C-language fprintf function)
<code>fprintfMat</code>	Write matrix to ASCII file under format control
<code>mfprintf</code>	Write data to ASCII file (C-type format)
<code>mprintf</code>	Writes data to Scilab window (C-type format)
<code>mput</code>	Write numeric data to file in user-specified binary representation
<code>mputl</code>	Write string vector to ASCII file (one line per vector element)
<code>mputstr</code>	Write character string to an ASCII file
<code>mtlb_save</code>	Save variables to file in Matlab-4 format
<code>print</code>	Print variables to file in the format used for Scilab window
<code>printf</code>	Print to Scilab window (emulation of C-language printf function)
<code>savewave</code>	Write a .wav sound file
<code>wavwrite</code>	Write a .wav sound file
<code>writb</code>	Write matrix in to a Fortran binary file (4 bytes/word)
<code>write</code>	Write matrix of strings/numbers to ASCII file (Fortran-type format)
<code>write4b</code>	Write matrix in to a Fortran binary file (8 bytes/word)

Table 5.10: Functions that output data to files or to the Scilab window

```

This is line 1
-->twomore = mgetl(fid,2) // Read two more lines
twomore =
!Line 2      !
!           !
!Line 3 (last) !
-->fclose(fid) // Close the file
ans =
0.

```

An analogous procedure can be used to write a file one line (or several lines) at a time.

It is important to note that `mputl` puts each string in a string matrix in a separate line. Thus a string matrix with more than one column — when read in — will become a one-column matrix. This is illustrated in the next example. 26a

```

-->textlines = ['This is line 1a','Line 1b';
-->             'This is line 2a','Line 2b']
textlines =
!This is line 1a Line 1b !
!                   !
!This is line 2a Line 2b !

```

```

-->mputl(textlines,'C:\temp\dummy.txt')

-->allnow = mgetl('C:\temp\dummy.txt') // Get the whole file
allnow =
!This is line 1a !
!
!This is line 2a !
!
!Line 1b !
!
!Line 2b !

```

The function `matrix` can be used to reshape (no pun on Matlab intended) `allnow` into the original 2 by 2 string matrix.

```

-->matrix(allnow,2,2) 27
ans =
! This is line 1a Line 1b !
!
! This is line 2a Line 2b !

```

It is a nice feature of `matrix` that only one of the dimensions needs to be given; the other can be replaced by -1. Thus statement 27 is equivalent to either of the two statements

```

matrix(allnow,-1,2)
matrix(allnow,2,-1)

```

The parameter not specified is computed from the dimension of the matrix to be reshaped.

### 5.3.3 Functions read and write

Functions `write` and `read` do what `mputl` and `mgetl` do — and more. The following statements are equivalent to those in 26a above.

```

-->textlines = ['This is line 1a','Line 1b';
-->            'This is line 2a','Line 2b']
textlines =
!This is line 1a Line 1b !
!
!This is line 2a Line 2b !
-->write('C:\temp\dummy.txt',textlines)

```

```

-->all = read('C:\temp\dummy.txt',-1,1,'(A)') // Get the whole file
  all =
! This is line 1a !
!                   !
! This is line 2a !
!                   !
! Line 1b         !
!                   !
! Line 2b         !

```

While the write statements only needs the file name and the data the read statement also wants the size of the array to read and a format in FORTRAN syntax. The dimension are in input arguments 2 and 3, the -1 simply instructs `read` to read the whole file; in this example it could have been replaced by 4 since there are 4 strings in the file. Like `mputl` function `write` writes a string array one column to a line.

Functions `read` and `write`, when used with a file name as first argument, open the file and close it again after the I/O operation. To read or write incrementally, one needs to open the file oneself. However, this cannot be done with function `mopen` used above. Rather, the file must be opened (and closed) with function `file`. This is illustrated in the example below where the file created above is read again.

```

-->fid = file('open','C:\temp\dummy.txt','unknown') // Open file
  fid =
    4.

-->one = read(fid,1,1,'(A)') // Read one line
  one =
    This is line 1a

-->twomore = read(fid,2,1,'(A)') // Read two more lines
  twomore =
! This is line 2a !
!                   !
! Line 1b         !

-->file('close',fid) // Close the file

```

Function `file` above opens the file `C:\temp\dummy.txt` for read and write access. By default the file is a sequential-access file for ASCII data. Other file types can be chosen by setting the appropriate input parameters.

Sequentially writing to a file is completely analogous.

Functions `read` and `write` can also be used to read and write numeric data.

```

-->fid = file('open','C:\temp\numeric.txt','unknown'); // Open file

-->a = rand(3,5,'normal')
a =
! - .7460990      .1023021  - .3778182  - .6453261  1.748736  !
! - 1.721103    - 1.2858605  2.5749104  .0116391  .1645912  !
! - 1.7157583    .6107784  - .4575284  - 1.4344473  .9182207  !

-->write(fid,a)

-->file('close',fid)                // Close the file

```

The 3 by 5 matrix **a** is written to file in ASCII format (as a string) and unformatted and can be retrieved as shown below.

```

-->[fid,ierr] = file('open','C:\temp\numeric.txt','old')
// Open file
ierr =
0.
fid =
4.

-->if ierr ~= 0 then error(' Problem opening file'), end

-->newa = read(fid,2,3)
newa =
! - .7460990      .1023021  - .3778182  !
! - 1.721103    - 1.2858605  2.5749104  !

-->file('close',fid)                // Close the file

```

Function **file** is used here with two output arguments; the second provides the error status. If an error occurs while opening a file function **file** does not abort but rather saves the error number in this second output argument and leaves it to the user to handle the error. Function **read** only requests the first two columns of the first two rows, and that is what is output. Furthermore, the file status is set to **'old'**. After all, the file must already exist in order to be read. Of course, **'unknown'** would have been an option too.

The next example shows how **a** can be written to a file under format control. It also shows that the “file” can be the Scilab window — as mentioned earlier, **%io(2)** is the file identifier for the Scilab window.

```

-->write(%io(2),a,'(5f10.5)') .51633 .64507 -.54852 -1.38505

```



```
-1.10499 1.04225 -.44840 1.13162 -1.62805 .76045 2.49761 -.72190
-1.36674 .77577 -.65881
```

Matrix **a** is written to the file in 5 columns, each of which is 10 characters wide, with 5 digits to the right of the decimal point. Incidentally, **write** can also be used to write a string vector (but not a matrix) to the Scilab window without the “almost blank” lines.

```
textlines(:)
ans =
!This is line 1a !
!
!This is line 2a !
!
!Line 1b      !
!
!Line 2b      !

-->write(%io(2),textlines)
This is line 1a
This is line 2a
Line 1b
Line 2b

-->write(%io(2),textlines,'(a20)')
      This is line 1a
      This is line 2a
          Line 1b
          Line 2b
```

Without a format the strings are left-justified. With format **a20** they are right justified; the total number of characters per line is 20.

### 5.3.4 Functions load and save

Functions **save** and **load** perform the same function they perform in Matlab: **save** writes one, or more, or even all variables of the workspace to a file. The file can be defined either by its name (in this case opening and closing is done automatically) or by a file identifier. In the latter case the file needs to be opened with **mopen** with parameter **wb** (write binary). But variables can be saved incrementally to the same file. An example is below.

```
-->a = 3;

-->fid = mopen('C:\Temp\saved.bin','wb');
```

```

-->save(fid,a)

-->b = 5; c = 'text';

-->save(fid,b,c)

-->fclose(fid);

```

Note that variable names in the `save` command are not in quotes. In Matlab they would be.

To recall variables saved earlier, possibly in another session,

```

-->clear a, clear b

-->load('C:\Temp\saved.bin','a','b')

-->a,b
a =
    3.
b =
    5.

```

Here, as in Matlab's `load` function, the variable names must be in quotes.

### 5.3.5 Functions `mput` and `mget/mgeti`

The two input functions allow one to read blocks of 1, 2, 4, or 8 bytes from a binary file and convert them into either double-precision floating point numbers (`mget`) or into integers (`mgeti`) (see rightmost column of the table below). The type of conversion is controlled by a type parameter which can take the following values

Type	in file	in Scilab
<code>c</code>	8-bit integer	<code>int8</code>
<code>s</code>	16-bit integer	<code>int16</code>
<code>i</code>	32-bit integer	<code>int32</code>
<code>l</code>	64-bit integer	<code>double</code>
<code>uc</code>	Unsigned 8-bit integer	<code>uint8</code>
<code>us</code>	Unsigned 16-bit integer	<code>uint16</code>
<code>ui</code>	Unsigned 32-bit integer	<code>uint32</code>
<code>ul</code>	Unsigned 64-bit integer	<code>double</code>
<code>f</code>	32-bit floating-point number	<code>double</code>
<code>d</code>	64-bit floating-point number	<code>double</code>

The functions can incrementally read/write files that have been opened with `mopen`.

With binary files the questions of byte ordering has to be addressed. Intel CPU's, and thus PC's, use "little-endian" byte ordering whereas so-called workstations (Sun Sparc, SGI, IBM RS/6000) use "big-endian" byte ordering. In Matlab byte ordering is specified when a file is opened with `fopen`. Function `mopen` in Scilab has no such option; instead, byte ordering is specified together with the variable type by appending a `b` or `l` to the type parameter. Thus the statement for reading 6 big-endian, 32-bit integers from a file with file identifier `fid` is

```
-->from_file = mget(6,'ib',fid)
```

Had the `b` been omitted the "natural" byte ordering of the computer on which the program runs would have been used (e.g. little-endian for a PC). As long as one reads files written on the same type of computer byte ordering is generally not a problem. It needs attention when one reads a file created on a computer with different byte ordering.

### 5.3.6 Functions input and disp

Function `input` is completely equivalent to Matlab's `input`:

```
-->response = input('Prompt user for input')
Prompt user for input-->3
response =
    3.
```

The user response (3 in this example) can also be an expression involving variables in the workspace. Furthermore, by adding a second argument, `'string'` or simply `'s'`, the user response can be interpreted as a string. There is no need to put it in quotes.

Function `disp` has a close Matlab analog as well. Unlike its Matlab counterpart it can take more than one argument. However, as illustrated out earlier (page 18) the arguments are displayed in reverse order.

### 5.3.7 Function xgetfile

Generally, functions whose name starts with an "x" have some connection to graphics (probably some reference to Xwindows). This is also true for `xgetfile` which opens a dialog box for interactive file selection. It works essentially like Matlab's `uigetfile`. An example is

```
-->file_name = xgetfile(filemask='*.sgy',dir='D:\Data\Seismic', ...
    title='Read SEG-Y file');
```

which opens a file selection window with the title "Read SEG-Y file". The initial directory shown in the window is `D:\Data\Seismic`, and only files with file name extension `.sgy` are shown initially.

## 5.4 Utility Functions

This chapter describes some of the functions that are not directly necessary to run or debug Scilab functions, that are more peripheral to Scilab and do not fit well in any other topic discussed previously. Table 5.11 shows the functions I chose to put into this category.

Scilab	Description
<code>diary</code>	Write screen output of a Scilab session to a file
<code>getversion</code>	Display version of Scilab
<code>host</code>	Execute Unix/DOS command; outputs error code
<code>lines</code>	Specify number of lines to display and columns/line
<code>pol2tex</code>	L <sup>A</sup> T <sub>E</sub> X representation of a polynomial
<code>stacksize</code>	Determine/set the size of the stack
<code>texprint</code>	L <sup>A</sup> T <sub>E</sub> X representation of a Scilab object
<code>timer</code>	Outputs time elapsed since the preceding call to <code>timer()</code>
<code>unix</code>	Execute Unix/DOS command; outputs error code
<code>unix_g</code>	Execute Unix/DOS command; output to variable
<code>unix_s</code>	Execute Unix/DOS command; no output (silent)
<code>unix_w</code>	Execute Unix/DOS command; output to Scilab window
<code>unix_x</code>	Execute Unix/DOS command; output to a new window

Table 5.11: Utility functions

Of the two functions that create L<sup>A</sup>T<sub>E</sub>X code `pol2tex` appears to be superfluous since it is more specialized and `texprint` does what it does (in fact, `pol2tex` produces sets of unnecessary braces).

The L<sup>A</sup>T<sub>E</sub>Xcode function `texprint` generates may require the `amsmath` package:

```
-->mat = [.2113249 .3303271 .8497452 .0683740 .7263507;
          .7560439 .6653811 .6857310 .5608486 .1985144;
          .0002211 .6283918 .8782165 .6623569 .5442573];

-->texprint(mat)
ans =
{\pmatrix{ .2113249& .3303271& .8497452& .068374& .7263507\cr
          .7560439& .6653811& .685731& .5608486& .1985144\cr
          .0002211& .6283918& .8782165& .6623569& .5442573}}
```

The `diary` function causes a copy of all subsequent keyboard input and the resulting Scilab output to be copied to the file in the argument. Is more limited than Matlab's `diary`, which only creates a new file if the named file does not exist. Otherwise, it appends the output to the existing file. Also, diary recording can be turned off and on. In Scilab the diary file always creates a new file and

aborts with an error message if the file already exists. Furthermore, `diary(0)` turns the recording off and closes the file. Recording cannot be toggled on and off.

Quite a few functions are available to execute UNIX or DOS commands from the Scilab command line. Those accustomed to the ways of Matlab will not be surprised to use a function that has `unix` in it to execute DOS commands. The choice among the various functions beginning with `unix` depends on the desired output which is indicated in Table 5.11. Functions `host` and `unix` are interchangeable.

Since operating system commands are generally different for DOS and UNIX, a function that is expected to run on both must have different branches for the two. The built-in variable `MSDOS` (see Table 4.6) can be used to determine the type of operating system.

```
-->if MSDOS
--> files=unix_g('dir D:\MyScilab\Experimental');
-->else
--> files=unix_g('ls -l ~/MyScilab/Experimental');
-->end
-->write(%io(2),files)

Volume in drive D has no label
Volume Serial Number is 18F2-2730
Directory of D:\MyScilab\Experimental

.                <DIR>          12-29-01  1:33p  .
..               <DIR>          12-29-01  1:33p  ..
READ_S 1 SCI      25,811  01-06-02  9:22p  read_segy_file.sci
READ_S 1 M        23,103  02-21-01  8:17p  read_segy_file.m
          2 file(s)          48,914 bytes
          2 dir(s)          858.80 MB free
```

## Chapter 6

# Scripts

A script is a sequence of Scilab commands stored in a file (while a script file may have any extension, the Scilab Group suggests the extension `.sce`). Scripts have neither input arguments nor output arguments. Since they do not create a new level of workspace all variables they create are available once the execution of the script is completed.

To invoke a script in Matlab its name without extension, say `script_file`, is typed on the command line. Furthermore, the file can be in any directory of the search path. Scilab, on the other hand, uses the concept of a working directory familiar from Unix. The command `pwd` (Print Working Directory) can be used to find out what it is. If a script, say `script_file.sce`, is in the working directory it can be executed by the command

```
-->exec('script_file.sce') 28a
```

A Scilab script can also be stored as a vector of strings; in this case it is executed by means of function `execstr`.

The function `exec` has two optional arguments: the string `'errcatch'` and the variable `mode`; the former allows a user to handle errors during execution of the script, the latter allows one to control the amount of output. It does not appear to really do what the documentation says. The following table is taken from the help file:

Value	Meaning
0	the default value
-1	print nothing
1	echo each command line
2	print prompt -- >
3	echo + prompt
4	stop before each prompt
7	stop + prompt + echo : useful mode for demos

The following are several examples of the mode parameters. Let `test.sce` be the following script

```
// Test script to illustrate the mode parameter
a = 1
b = a+3;
disp('mode = '+string(mode()))
```

Then, without setting the mode parameter, i.e.  
**mode not specified:**

```
-->exec('D:\MyScilab\test.sce')

-->// Script to explain mode parameter

-->a=1
a =

    1.

-->b=a+3;

-->disp('mode = '+string(mode()))

mode = 3

-->disp('mode = '+string(mode()))

mode = 2
```

In this case **exec** echoes every line of the script and displays the results of every statement without a terminating semicolon. Here and in the following examples spaces between lines output by **test** have been preserved to more accurately reflect the output of the script. Obviously, the default is for **exec** to set **mode** to 3. But once **exec** has run **mode** reverts to 2.

**mode = 0:**

```
-->exec('D:\MyScilab\test.sce',0)
a =

    1.

mode = 0
```

This value of mode produces the result one would expect from Matlab.

**mode = 1:**

```
-->exec('D:\MyScilab\test.sce',1)
```

```

-->// Script to explain mode parameter
-->a = 1
  a =

      1.
-->b = a+3;
-->disp('mode = '+string(mode()))

mode = 1

```

This is the same information displayed with `mode = 0` but in a somewhat more compact form (fewer blank lines).

**mode = -1:**

```

-->exec('D:\MyScilab\test.sce',-1)

mode = -1

```

In this case the result of expressions is not displayed even if they are not terminated by a semicolon.

**mode = 2:**

Displays the same information as `mode = 1` but with more empty lines.

**mode = 3:**

Default mode (mode parameter not given).

**mode = 4:**

Prints `step-by-step mode: enter carriage return to proceed` but then behaves like `mode = 0`.

**mode = 7:**

```

-->exec('D:\MyScilab\test.sce',7)
step-by-step mode: enter carriage return to proceed
>>
-->// Script to explain mode parameter
>>
-->a = 1
  a =

      1.
>>
-->b = a+3;
>> -->>disp('mode = '+string(mode()))

mode = 7

```



This mode works as advertised. It prompts the user for a <RETURN> after each statement. `mode` Function `exec` satisfies the requirements of the command-style syntax (see Section 5.1). Thus `28a` and `28b`, `28c` below are equivalent statements.

```
-->exec 'script_file.sce' 28b
```

and

```
-->exec script_file.sce 28c
```

Furthermore, for all three variants, a trailing semicolon will suppress echoing the commands `exec` executes.

As in Matlab, Scilab scripts can include function definitions. However, Scilab is more flexible in the way functions can be defined within a script (or within another function). This is explained below in Section 7.2.

If a file with a Scilab script is not in the working directory then either the working directory needs to be changed (with function `chdir`) or the full filename of the script must be given. Scilab does not provide for a search path the way Matlab does or the way Unix provides for executables.

A bare-bones simulation of a search path for the execution of a Scilab script is afforded by the following function.

```
function myexec(filename,mod)
// Function emulates use of a search path for the execution of a script
//
// INPUT
// filename filename of the script to be executed; the
//          extension .sce is added if the file name
//          of the script has no extension
// mod      mode parameter(determines amount of printout;
//          see help file for exec); default: mod = 1.
//
// EXAMPLES: myexec('test')
//          myexec test 1

global PATH

if PATH == [] // Set search path if it is not defined globally
    path=['D:\MyScilab\Experimental\ ', 'D:\MyScilab\tests\ ', ...
        'D:\MyScilab\General\ ', 'D:\MyScilab\Sci_files\'];
else
    path=PATH;
```

```

end

if argn(2) == 1      //Set default for mod, if not given
    mod=0;
end

//      Add extension to file name if necessary
tempname = filename;
lf=length(tempname);
if lf > 3
    if part(tempname,lf-3:lf) ~= '.sce'
        tempname=tempname+'.sce';
    end
else
    tempname=tempname+'.sce';
end

//      Test the filename with directories in path
for ii=1:size(path,'*');
    testfile = path(ii)+tempname;
    [fid,ierr]=file('open',testfile,'old');

    if ierr == 0
        file('close',fid)
        oldvars=who('local');      // Variables before execution of script

        exec(testfile,mod);      // Execute the script

//      Return variables created in script to the level from which
//      myexec was called
        newvars=who('local');      // Variables after execution of script
        newvars=newvars(1:size(newvars,'*')-size(oldvars,'*')-1);
        str1 = strcat((newvars+',')');
        str1 = part(str1,1:length(str1)-1); // Comma-separated list of
            // variables to return to lower level
        execstr([''+str1+']=return(''+str1+')') // Return to lower level
    end

    if ierr ~= 240
        break
    end
end
end

```

```
if ierr == 240 // File not found in any of the directories
    write(%io(2), 'File '+tempname+ ' not found. Directories searched:')
    write(%io(2), '    '+path)
end
endfunction
```

The four directories of the search path are defined in the string vector `path`. One directory after the other is concatenated with the file name `filename` of the script to be executed. If a file by this name does not exist in the folder then `file` aborts with error 240 (File filename does not exist or read access denied) and the next directory is tried. If `file` is successful it is closed again and `exec` is executed with the file in that directory. In the next step any variables that have been created by the script are returned to the calling program.

If the file is in none of the directories the function prints an error message, lists the directories in the search path, and terminates.

## Chapter 7

# User Functions

While functions in Scilab are variable and not files they have many features in common with those in Matlab. They consist of a function head and the function body. The function head has the form

```
function [out1,out2,...] = function_name(in1,in2,...)
```

familiar from Matlab. The `...` indicate that the number of input and output arguments is arbitrary. A function may have no input and/or no output arguments. For functions with one output argument, the brackets are optional. For functions with no input arguments the parentheses are optional when it is defined, but not when it is called. No characters other than white spaces — not even comments — are allowed after the closing parenthesis. The function head can extend over more than one line with the usual continuation indicator (`...`).

The function body consists of a number of Scilab statements. Functions can be either in separate files (one or more functions per file and the name of the file is not necessarily related to the names of the functions) or they can be created within scripts or other functions (in-line functions). Functions can be used recursively, i.e. a function can call itself.

An example of a simple function is

```
function [r,phi] = polcoord(x,y)
    // Function computes polar coordinates from cartesian coordinates
    r = sqrt(x^2+y^2);
    phi = atan(y,x)*180/%pi;
endfunction
```

This function looks much like a Matlab function except for:

- the two slashes preceding the comment;
- the use of the function `atan` rather than its Matlab equivalent `atan2`;
- the use of special constant `%pi` rather than its Matlab equivalent `pi`

- the use of `endfunction` which can be omitted (but would be required if `polcoord` were defined in-line).

Scilab	Description
<code>argn</code>	Number of input/output arguments of a function
<code>endfunction</code>	Indicate end of function
<code>error</code>	Print error message and abort
<code>function</code>	Identify header of function definition
<code>halt</code>	Stop execution and wait for a key press
<code>mode</code>	Control amount of information displayed by function/script
<code>pause</code>	Interrupt execution of function or script
<code>resume</code>	Return from a function or resume execution after a <code>pause</code>
<code>return</code>	Return from a function or resume execution after a <code>pause</code>
<code>varargin</code>	Variable number of input arguments for a function
<code>varargout</code>	Variable number of output arguments for a function
<code>warning</code>	Print warning message
<code>where</code>	Output current instruction calling tree
<code>whereami</code>	Display current instruction calling tree
<code>whereis</code>	Display name of library containing function

Table 7.1: Functions/commands/keywords relevant for user functions

The following example computes the Chebyshev polynomial  $T_n(x)$  by means of the recurrence relation

$$T_{n+1}(x) = 2xT_n(x) - T_{n-1}(x)$$

to illustrate the recursive use of functions (functions calling themselves).

```
function ch = cheby(x,n)
// Compute Chebyshev polynomial of order n for argument x
if n == 0
    ch = 1;
elseif n == 1
    ch = x;
else
    ch = 2*x*cheby(x,n-1)-cheby(x,n-2);
end
endfunction
```

In Scilab, variables can be passed to functions in three different ways:

- as a variable in the input argument list

- as a global variable
- as a variable not local to the function, i.e. a variable that is not initially defined in the function

The first two ways of input and output are also used by Matlab. The third one is not. It essentially means that any variable defined in the calling workspace of a function is available to that function as long as it is not defined there. Even if the statement `endfunction` were omitted the function

```
function y = func1(x) 30a
    a = (a+1)^2
    y = x+a;
endfunction
```

would not work in Matlab since the variable `a` is not defined prior to its first use in `func1`. In Scilab the following code fragment works:

```
-->a = 1; 31a

-->y = func1(3)
y =
    7.

-->disp(a)
    1.
```

Since the variable `a` (set to 1) is available in the calling workspace, it is also available in `func1`. The new value of `a` created in `func1` (`a` is changed to 4) is not passed on to the calling workspace. This approach works across an arbitrary number of levels. Assume `funcB(x)`, which uses a variable `a` without first defining it, is called by `funcA(x)` which does not use a variable `a`. Then

```
a = 5; funcA(10);
```

still works: `a` in `func2B` is taken to be 5 since `a` is part of the calling workspace not only of `funcA` but also of `funcB`. So one might wonder about the purpose of the `global` statement if variables are passed to functions even if they are not in the argument list or defined as global. The answer is simply that the global statement allows one to "export" variables from a function. Thus changing line 31a by defining `a` as global has no effect on the result

```
-->global a, a = 1; 31b

-->y = func1(3)
y =
```

7.

```
-->disp(a)
1.
```

However, if function `func1` [30a] is changed to `func1g` which also includes a global statement

```
function y = func1g(x) [30b]
    global a
    a = (a+1)^2
    y = x+a;
endfunction
```

then

```
-->global a [31c]

-->a = 1;

-->y = func1g(3)
y =
    7.

-->disp(a)
4.
```

The variable `a` at the end of code fragment [31c] is 4, the value computed in `func1g`. If the `global a` is dropped from [31c] then

```
-->a = 1; [31d]
-->y = func1g(3)
y =
    7.
-->disp(a)
1.
```

Thus [31a], which uses `func1`, and [31d], which uses `func1g`, leave the variable `a` unchanged in the calling program where it is not defined as global.

Scilab — like Matlab — has variable-length input argument and output argument lists. They even have the same names, `varargin` and `varargout`, and work the same way. If specified together with regular (positional) arguments, they must be last in the argument list. An example is

```

function sizes(varargin)
    // Arguments must be numeric or string matrices
    for ii = 1:length(varargin)
        [nrows,ncols] = size(varargin(ii));
        disp('Input argument no '+string(ii)+' has '+string(nrows)...
            +' row(s) and '+string(ncols)+' column(s)')
    end
endfunction

```

which can be called with any number of input arguments and prints the number of rows and columns for each input argument (provided the input arguments are not lists and the like for which `size` has fewer than 2 or more than 2 output arguments). Thus

```

-->sizes(1:10,'test',['a','ab';'abc','abcd'])
Input argument no 1 has 1 row(s) and 10 column(s)
Input argument no 2 has 1 row(s) and 1 column(s)
Input argument no 3 has 2 row(s) and 2 column(s)

```

The number (`in`) of actually defined input arguments and the number (`out`) of output arguments of a function is provided by function `argn` as follows

```

[out [,in] ]=argn()
out=argn(1)
in=argn(2)

```

This function does what `nargin` and `nargout` do in Matlab. However, there is a slight twist. It is not possible to determine if a function has been called without an explicit output argument (left-hand side) since there is always the implied left-hand side `ans`. Thus `argn(2)` will never be 0.

In Scilab variables can be output in three different ways as well:

- as a variable in the output argument list
- as a global variable
- as the argument of the `resume` or `return` command

The first two are familiar from Matlab; furthermore, the above discussion of global variables has also touched on the role of global variables as means to output data from a function. So it is only the third item that needs an explanation.

In order to explain how the third way of returning parameters works it is necessary discuss a difference between the Matlab `keyboard` command and the Scilab `pause` command. Both commands interrupt the execution of a function or script. In Matlab one ends up in the workspace of the



interrupted function (or script). Any variable created in this workspace is available to the interrupted function once execution resumes. Scilab, on the other hand, creates a new workspace. As with functions, all variables defined in the higher workspaces are available. But, upon return to the workspace above (Scilab command `resume`, all newly created variables (or any modifications of variables of the higher workspaces) are not available to this workspace.

Before I go on to the next section it is appropriate to shed some light on this section's opening statement that functions in Scilab are variables. The following sequence of Scilab statements is meant to illustrate this somewhat abstract statement.

```

-->a = 1;

-->typeof(a)
ans =
constant

-->convstr('AbCdE')
ans =
abcde

-->typeof(convstr)
ans =
function

-->a = convstr;

-->typeof(a)
ans =
function

-->a('UvWxY')
ans =
uvwxy

```

Initially, the variable `a` is assigned the value 1 and is of type `constant`. On the other hand, the function `convstr`, which converts upper case characters to lower case, is of type `function`. Obviously, like `a`, `convstr` is used as an argument of function `typeof`. Now I set `a` equal to `convstr` (note, that `convstr` is used without parentheses or argument). This turns `a` into a function and, as shown in the last statement, makes it an alias for `convstr`.

The fact that functions are variables has number of advantages not the least of which is that they can input or output arguments of other functions (in Matlab one needs to pass the function name as a string and use `feval` to evaluate it).

## 7.1 Functions in Files

Scilab does not come with an integrated text editor. Thus files with Scilab functions are usually created with a text editor of the user's choosing. For Windows, PFE is a good choice, for Unix systems it is frequently `emacs` or `nedit`. The former is the default for the `edit` command (which is not implemented in the Windows version). Scilab function files generally have the extension `.sci` though, in principle, any other extension or no extension could be used as long as these files are loaded into Scilab with the `getf` command. However, some functions, like `genlib` and `getd` use the extension to identify files with Scilab functions. Hence, it is a good idea to comply with this convention.

Like in Matlab, there can be more than one function in a file. But in Matlab the file name is actually the function name, and the second, third, etc. function in a file is only visible to the first function. In Scilab the file name is immaterial and all functions in a file are “visible” to any function loaded, command-line statement, or script.

In order to be able to use a function defined in a file it has to be loaded first (this means a significant departure from the approach Matlab uses where the interpreter searches the directories of the search path for the function and loads and compiles the function in the first file encountered with the name).

A function can be loaded via the `getf('filename')` command. Here `filename` is the name of the file containing the function; if this file is not in the working directory the full path must be given. Thus

```
-->getf('polcoord.sci')
```

is sufficient to load the file `polcoord.sci` if it is in the working directory. If this is not the case then something like

```
-->getf('D:\MyScilab\Filters\polcoord.sci')
```

must be used. It is important to note that the filename must include the extension (whereas Matlab implies the extension `.m`). Once `getf` is executed the functions in the file are immediately available. This differs from the `load` command discussed below. Also, see the “gotcha” regarding `getf` on page 101.

Functions can also be collected in libraries. This is discussed in Chapter 8. It is important to remember, however, that loading a library does not mean that the functions in it are loaded but rather that they are marked as available to be loaded when called—**provided** they are undefined at that time. If the name of a function happens to be that of an already defined function or a built-in function it will never be loaded. One can use `getf` to force loading of a function (provided it does not have the same name as a protected built-in function).

This last condition points to one of the challenges of writing a function: choosing its name. It is important that a name reflects the purpose of the function, is easy to remember, and is not already

used. The set of names that satisfy these criteria is surprisingly small — significantly smaller than in Matlab. And there are three reasons:

1. Variable names are shorter (24 vs 31 characters in Matlab)
2. In Matlab a second, third, etc. function in a single file is only visible to the first function in the file. So there is no conflict with any other function in Matlab’s search path.
3. Matlab has the concept of “private functions”. These are functions that reside in a subdirectory named `private` and that are only visible to functions in the parent directory: when a function in a directory that has a subdirectory `private` calls another function the subdirectory `private` is searched first to check if the function is there; only if it is not found the standard search path is checked.

It is particularly the lack of the “private directory” concept that makes writing a program package that peacefully coexists with other packages more challenging than it needs be.

## 7.2 In-line Functions

Functions need not be set-up in files. They can also be created “on the fly”. There are two ways to do so; one of them has already been used for examples in previous chapters (see e. g. function `ismember` on page 39). The function can be typed into the Scilab window as if it were typed in a file editor; the important thing to remember is that the statement `endfunction` is required to tell the interpreter that the function definition is complete. While the function statements are typed in, the usual double-spaced display format is replaced by single spacing.

The other way of inputting a function uses the function `deff`. A simple example of its use is

```
-->deff('y = funct(x)', 'y = x^2')

-->funct(3.5)
ans =
    12.25

-->typeof(funct)
ans =
function

-->type(funct)
ans =
    13.
```

The first argument of `deff` is a character strings with the function header, the second is a string or a string vector which contains the body of the function. An optional third argument specifies if

the function should be compiled `'c'` or not `'n'`. The former is more efficient than the latter and is the default. Thus, in the example above, `funct` is compiled. This is also proven by the fact that `funct` has type 13 (see Table 4.1). On the other hand, with

```
-->deff('y = funct(x)', 'y = x^2', 'n');

-->typeof(funct)
ans =
function

-->type(funct)
ans =
11.
```

The variable `funct` has type 11 (uncompiled function), while the output of `typeof` is unchanged.

With more complicated functions or functions that contain string definitions the first version of in-line function definition is generally easier to read.

It is important to note that inline functions can be defined not just in the command window. They can also be included in Scilab scripts and functions.

### 7.3 Functions for operator overloading

Operator overloading refers to the ability to give operators that are used for one kind of data object a new meaning for another one. An example mentioned before is the use of the `+` to concatenate two strings or string matrices. But not only operators can be overloaded; The way a data object is displayed can be overloaded as well. For typed lists and matrix-oriented typed lists it is the type name, the first string in the first entry of a typed list, that defines the type of data object. The typed list `seismic_data` has type `seismic`; it simulates a seismic data set with 10 seismic traces, each consisting of 251 samples; hence, in the following, it is generally referred to as “seismic data set”

```
-->seismic_data = tlist(['seismic', 'first', 'last', 'step', ...
                        'units', 'traces'], 0, 1000, 4, 'ms');

-->nsamp = (seismic_data.last - seismic_data.first) / seismic_data.step + 1;

-->seismic_data.traces = rand(nsamp, 10);

-->seismic_data

seismic_data(1)
```

```

!seismic first last step units traces !

    seismic_data(2)
0.

    seismic_data(3)
1000.

    seismic_data(4)
4.

    seismic_data(5)
ms

    seismic_data(6)
    column 1 to 5
!   .3914068   .2173720   .4883297   .4061224   .9985317 !
!   .8752304   .4418458   .9141346   .9613220   .1959695 !
!   .5266080   .9798274   .6645192   .8956145   .9872472 !
!   .9856596   .5259225   .5468820   .0717050   .4248699 !
[More (y or n ) ?]

```

The default display of such a typed list is needlessly long; for this reason the function `show` had been introduced to provide a more compact display for typed lists (see page 44). It would be highly desirable to use `show` as the default display of typed lists of type `seismic`. This can be done surprisingly easily by means of the function

```

function %seismic_p(seis)
// Function displays the typed list seis of type 'seismic' much like
// a Matlab structure
    show(seis)
endfunction

```

The result is

```

-->seismic_data =

LIST OF TYPE "seismic"
    first: 0
    last: 1000
    step: 4
    units: ms
    traces: 251 by 10 matrix

```

Operator	Op-code	Operator	Op-code
'	t	\.	w
+	a	[a,b]	c
-	s	[a;b]	f
*	m	() extraction	e
/	r	() insertion	i
\	l	==	o
^	p	<>	n
.*	x		g
./	d	&	h
.\	q	.^	j
.*.	k	~	5
./.	y	.'	0
.\.	z	<	1
:	b	>	2
*.	u	<=	3
./.	v	>=	4

Table 7.2: Operator codes used to construct function names for operator overloading

Overloading the way a variable is displayed is possible because the typed list `seismic` looks for a function with the name `%seismic.p`. As shown in this example the function name consists of the `%` sign followed by the type name, `seismic`, an underscore as a separator, and the letter `p` which indicates display (the fact that the underscore serves as a separator between the list type and the “p” does not mean that there cannot be an underscore in the type name).

In principle, any operator that is not predefined for given types of operands can be overloaded. The name of the overloading function is constructed according to certain rules. For unary operators (`-`, `'`, and `~`), for example, it has the form `%<operand_type>_<op_code>`. An example is the use of the minus sign in front of the seismic-typed list `seismic_data` to change the sign of the entries of the matrix `seismic_data.traces`. As shown in Table 7.2 the operator code for the minus sign is `s`. Thus

```
function seismic = %seismic_s(seismic)
// Function defines the unary negation for a seismic structure
    seismic.traces=-seismic.traces;
endfunction
```

With the seismic data set `seismic_data` defined above

```
-->seismic_data.traces(1,1)
ans =
    .2113249
```

```

-->seismic_data = -seismic_data;

-->seismic_data.traces(1,1)
ans =
- .2113249

```

The function name for overloading binary operators has the form %<first\_operand\_type>.<op\_code>.<second\_operand\_type>. In this definition <operand\_type> is code for the type of variable the operator of type <op\_code> is operating on. Operand types, i.e. codes for the various Scilab variables, are listed in the rightmost column of Table 4.1 on page 19. An example is the following function which defines the operation of adding a scalar to a seismic data set.

```

function seismic = %seismic_a_s(seismic,c)
// Function adds a constant to the matrix "seismic traces"
    seismic.traces = seismic.traces + c;
endfunction

```

Here <first\_operand\_type> is `seismic` and the <second\_operand\_type> is `s`. A quick look at Table 4.1 shows that `s` is the operand type of a constant. As shown in Table 7.2, `a` is the operator code for `+` (addition). Thus

```

-->seismic_data.traces(1,1)
ans =
.2113249

-->seismic_data = seismic_data + 1;

-->seismic_data.traces(1,1)
ans =
1.2113249

```

It is important to note that overloading the operator `+` via `%seismic_a_s(seismic,c)` is only defined for this specific sequence of operands. The expression `1 + seismic_data` causes an error message as does, for example, `seismic_data - 1`; but `seismic_data + (-1)` works.

Some primitive functions can also be overloaded if they are not defined for the data type. In this case the function name has the form %<type\_of\_argument>.<function\_name>. The function below takes the absolute value of the traces of a seismic data set.

```

function seismic = %seismic_abs(seismic)
// Function takes the absolute value of the entries of the
// matrix 'seismic.traces'
    seismic.traces = abs(seismic.traces);
endfunction

```

Thus, for the seismic data set `seismic_data` defined above,

```
-->seismic_data.traces(1,1)
ans =
    .2113249

-->seismic_data = abs(-seismic_data);

-->seismic_data.traces(1,1)
ans =
    .2113249
```

Extraction of object elements can be overloaded by means of a function the name of which has the form `%<operand_type>_e(i1,...,in,operand)`. A somewhat simplified example is

```
function seis = %seismic_e(i,j,seis)
    // Function extracts rows i and columns j of the ...
    // matrix "seis.traces"; i and j can be vectors
    seis.traces = seis.traces(i,j);
    seis.last = seis.first+(i($)-1)*seis.step;
    seis.first = seis.first+(i(1)-1)*seis.step;
endfunction
```

which outputs a seismic data set where `seis.matrix` consists only of the elements `i,j` of the input matrix. An example is

```
-->seismic_data.traces(5,10)
ans =
    .1853351

-->temp = seismic_data(5,10)

temp =

LIST OF TYPE "seismic"
    first: 16
    last: 16
    step: 4
    traces: .1853351
    units: ms
```

It is important to keep in mind that typed lists have extraction (component extraction) defined for one index. Hence, extraction with one index cannot be overloaded, and



```
-->seismic_data(4)
ans =
    4.
```

produces the fourth element of typed list `seismic`, `step`, which is 4 (remember that the first element is the string vector `['seismic', 'first', 'last', 'step', 'traces']`).

It is also possible to extract data object elements to more than one output objects. Furthermore, the insertion syntax and row and column concatenation can also be overloaded.

## 7.4 Translation of Matlab-4 m-files to Scilab Format

A function, `mfile2sci`, is available to translate Matlab m-files to Scilab. This function is still being worked on by someone in the Scilab team (it is likely to be a never-ending task) but the existing version greatly simplifies this kind of conversion. It relieves the user of a lot of drudgery and lets him concentrate on the thornier problems: instances where Matlab and Scilab functions may differ slightly, possibly depending on parameters in the argument list, where functions unknown to `mfile2sci` are used, etc. `mfile2sci` allows individual files or whole directories to be converted. In the process it creates a \*.sci file, a “compiled” \*.bin file, and a \*.cat file (help file). The latter is generated from the comment lines at the beginning of the m-file, those lines that are also used by Matlab’s help facility. An example is

```
mfile2sci('D:\MyScilab\Geophysics\read_las_file.m', ...
         'D:\MyScilab\Geophysics')
```

Details about the conversion can be found in file `SCIDIR\macros\m2sci\README`; `SCIDIR` denotes the Scilab root directory (in Windows something like `C:\Program Files\Scilab`).

Another useful function is `translatepaths` which translates all Matlab m-files in a set of directories to Scilab. It uses `mfile2sci` to translate the individual files.

## Chapter 8

# Function Libraries

This is a topic that has no analog in Matlab. Libraries are collections of compiled functions that can be loaded automatically upon startup or that can be loaded on demand. There are several ways of creating libraries; the one described in the following appears to be the least painful.

Say, `C:\MyScilab` is a directory/folder with two Scilab functions (file extension `.sci`).

```
-->unix_w('ls C:\MyScilab')
lower.sci
upper.sci
```

Then a possible procedure to create a library is as follows;

```
-->genlib('Mylib','C:\MyScilab')
```

Function `genlib` compiles every Scilab function (file with extension `.sci`) in directory `C:\MyScilab` and saves it in a file with the same root but extension `.bin`. It also creates the text file `names` with the names of all functions, and a library file `lib`. Hence, directory `C:\MyScilab` now contains the following files

```
-->unix_w('ls C:\MyScilab')
lib
lower.bin
lower.sci
names
upper.bin
upper.sci
```

In addition the variable `Mylib` of type library (type 14) is created in the workspace and all Scilab functions in `C:\MyScilab` are now available for use.

It is important to note that this does not create help files. This has to be done separately. Furthermore, the statement

```
load('D:\MyScilab\lib');
```

in the startup file `.scilab` will load the library `Mylib` every time Scilab is started. Note that the library name `Mylib` is not mentioned in the load statement. Nevertheless, the variable `Mylib` of type library shows up in the workspace (the library `lib` “knows” that its name in the workspace is `Mylib`).

There are a few things to keep in mind with regard to loading libraries. This is illustrated in the following.

```
-->clear // Remove all unprotected variables from the workspace

-->who // Show all variables
your variables are...
%helps scicos_pal MSDOS home PWD TMPDIR
plotlib percentlib soundlib xdesslib utillib tdcslib
siglib s2flib roplib optlib metalib elemllib commlib
polylib autolib armalib alglib intlib mtlbllib WSCI
SCI %F %T %z %s %nan %inf
$ %t %f %eps %io %i %e
using 5517 elements out of 10000000. and 41
variables out of 1791

-->lc = lower('ABC')
!--error 4
undefined variable : lower
```

Since all unprotected variables have been removed the function `lower` is not available. To get it back one can load the library `Mylib`, and the command `who` shows that the variable `Mylib` is now in the workspace.

```
-->load('D:\MyScilab\lib') // Load library containing function lower

-->who
your variables are...
Mylib %helps scicos_pal MSDOS home
PWD TMPDIR plotlib percentlib soundlib
xdesslib
utillib tdcslib siglib s2flib roplib optlib metalib
elemllib commlib polylib autolib armalib alglib intlib
mtlbllib WSCI SCI %F %T %z %s
%nan %inf $ %t %f %eps %io
%i %e
```

```

using      5553 elements out of  10000000.
and        42 variables out of    1791

```

It is important to note, however, that loading a library does not mean that the functions in it are loaded into the workspace; they are only marked as available to be loaded when called. Hence, `lower` is not listed yet. Nevertheless, we can now execute the function `lower` and expect it to be loaded.

```

-->lc = lower('ABC')
lc =
abc

-->who
your variables are...
lc      lower      Mylib      %helps      scicos_pal
MSDOS   home          PWD        TMPDIR      plotlib     percentlib
soundlib xdesslib      utillib    tdcslib     siglib      s2flib     roplib
optlib  metalib      elemllib   commlib     polylib     autolib    armalib
alglib  intlible     mtlblib    WSCI        SCI         %F         %T
%z      %s           %nan       %inf        $           %t         %f
%eps    %io         %i         %e
using   5650 elements out of  10000000.
and     44 variables out of    1791

```

This statement adds two more variables to the workspace: the string variable `lc` and the function `lower`.

Functions in libraries are actually loaded only if they are still undefined and their name is encountered during execution! Thus a potential problem exists if the library function name is the same as that of a built-in function or an already defined user function. In this case it would not be loaded. A related problem would come up if one found a bug in, say, `lower`, fixed it, and rebuilt and reloaded the library. If `lower` is executed again one would not get the corrected version in the rebuilt library `Mylib` but rather the one in variable `lower`. Hence, in order to get the corrected version it is not only necessary to rebuild and load the new library but also to remove the variable `lower` from the workspace; in other words: it is necessary to execute the command

```

-->clear lower

```

An alternative is to bring the corrected version of `lower` into the workspace via

```

-->getf('C:\MyScilab\lower.sci')

```

# Chapter 9

## Gotchas

This chapter deals with unexpected problems I encountered during my travails.

### Function `getf`

Function `getf` reads and compiles a function from a file (see page 90). In case it encounters an error while compiling it aborts with an error message. When one corrects the error and wants to save the file to repeat the call to `getf` one finds out that this is not possible since `getf` has not closed the file. The sledge-hammer approach to this problem is to close all user files with `fclose all`. A more nimble approach is to find the offending file's identifier by executing `dispfiles()` and then close only that specific file. This is illustrated below

```
-->dispfiles()
|File name                                |Unit|Type|Options          |
|-----|-----|-----|-----|
|D:/PROGRAM FILES/SCILAB-2.6/scilab.hist |1   |F77 |unknown formatted |
|D:\MyScilab\Tests\read_seggy_file.sci   |2   |C   |r b                |
|Input                                    |5   |F77 |old formatted      |
|Output                                    |6   |F77 |new formatted      |

-->fclose(2);
```

### Line numbers in error messages

Line numbers displayed with error messages all too frequently do not agree with the line numbers of the offending statement in the function file. Apparently, there are various reasons for that. If there are comment lines prior the function header those lines are not counted. Other irregularities seem to be associated with expressions that continue over two or more lines.

## Appendix A

# Matlab functions and their Scilab Equivalents

The following table is an alphabetic list of Matlab functions and their Scilab functional equivalents. The third column contains one-line descriptions that pertain to the Scilab function and not to the Matlab function (in case there is a difference). In some instances the term "equivalent" is defined rather loosely; parameters may be different or the output may be somewhat different in certain circumstances (an example is the Scilab function `length` which for numeric matrices or string matrices produces a different result than the Matlab function `length`). In other cases functions provide the same functionality, but in a somewhat different way. For this reason it is not generally sufficient to replace a Matlab function by the equivalent listed here; it is necessary to check the Scilab help file before using one of these equivalents.

Matlab	Scilab	
<code>[]</code>	<code>[]</code>	Empty matrix
<code>abs(a)</code>	<code>abs(a)</code>	Absolute value of <code>a</code> , $ a $
<code>acos</code>	<code>acos</code>	Arc cosine
<code>acosh</code>	<code>acosh</code>	Inverse hyperbolic cosine
<code>all</code>	<code>and</code>	Logical AND of the elements of boolean or real numeric matrix <code>a</code>
<code>all(a)</code>	<code>and(a)</code>	Output <code>%t</code> if all entries of the boolean matrix <code>a</code> are true
<code>any</code>	<code>or</code>	Logical OR of the elements of boolean or real numeric matrix <code>a</code>
<code>asin</code>	<code>asin</code>	Arc sine
<code>asinh</code>	<code>asinh</code>	Inverse hyperbolic sine
<code>atan</code>	<code>atan</code>	Arc tangent
<code>atan2</code>	<code>atan</code>	Arc tangent
<code>atanh</code>	<code>atanh</code>	Inverse hyperbolic tangent
<code>balance</code>	<code>balanc</code>	Balance matrix to improve condition number

Matlab	Scilab	
<code>besseli</code>	<code>besseli</code>	Modified Bessel function of the first kind, $I_\alpha(x)$
<code>besselj</code>	<code>besselj</code>	Bessel function of the first kind, $J_\alpha(x)$
<code>besselk</code>	<code>besselk</code>	Modified Bessel function of the second kind, $K_\alpha(x)$
<code>bessely</code>	<code>bessely</code>	Bessel function of the second kind, $Y_\alpha(x)$
<code>break</code>	<code>break</code>	Force exit from a <code>for</code> or <code>while</code> loop
<code>case</code>	<code>case</code>	Start clause within a <code>select</code> block
<code>ceil(a)</code>	<code>ceil(a)</code>	Round the elements of <code>a</code> to the nearest integers $\geq a$
<code>char</code>	<code>ascii</code>	Convert ASCII codes to equivalent string
<code>chol(M)</code>	<code>chol(M)</code>	Choleski factorization; $R'R = M$
<code>clear</code>	<code>clear</code>	Clear unprotected variables and functions from memory
<code>clear global</code>	<code>clearglobal</code>	Clear global variables from memory
<code>compan</code>	<code>companion</code>	Companion matrix
<code>cond</code>	<code>cond</code>	Condition number of <code>M</code>
<code>cond</code>	<code>cond</code>	Condition number of a matrix
<code>conj</code>	<code>conj</code>	Complex conjugate
<code>conv</code>	<code>convol</code>	Convolution
<code>cos</code>	<code>cos</code>	Cosine
<code>cosh</code>	<code>cosh</code>	Hyperbolic cosine
<code>cot</code>	<code>cotg</code>	Cotangent
<code>coth</code>	<code>coth</code>	Hyperbolic cotangent
<code>cumprod</code>	<code>cumprod</code>	Cumulative product of all elements of a vector or array
<code>cumsum</code>	<code>cumsum</code>	Cumulative sum of all elements of a vector or array
<code>date</code>	<code>date</code>	Current date as string
<code>dbstack</code>	<code>whereami</code>	Display current instruction calling tree
<code>dbstack</code>	<code>where</code>	Output current instruction calling tree
<code>deblank</code>	<code>stripblanks</code>	Remove leading and trailing blanks from a string
<code>det</code>	<code>det</code>	Determinant of a matrix
<code>diag</code>	<code>diag</code>	Create diagonal matrix or extract diagonal from matrix
<code>diary</code>	<code>diary</code>	Write screen output of a Scilab session to a file
<code>disp</code>	<code>disp</code>	Display input argument
<code>double</code>	<code>double</code>	Convert integer of any type/length to floating point
<code>double</code>	<code>ascii</code>	Convert string to equivalent ASCII codes
<code>echo</code>	<code>mode</code>	Control amount of information displayed by function/script
<code>eig</code>	<code>bdiag</code>	Block diagonalization of matrix
<code>eig</code>	<code>spec</code>	Eigenvalues of matrix
<code>ellipj</code>	<code>%sn</code>	Jacobian elliptic function, <code>sn</code>
<code>else</code>	<code>else</code>	Start an alternative in an <code>if</code> or <code>case</code> block

Matlab	Scilab	
<code>elseif</code>	<code>elseif</code>	Start a conditional alternative in an <code>if</code> block
<code>end [loop]</code>	<code>end</code>	Terminate <code>for</code> , <code>if</code> , <code>select</code> , or <code>while</code> clause
<code>end [matrix]</code>	<code>\$</code>	Index of last element of matrix or (row/column) vector
<code>erf</code>	<code>erf</code>	Error function, $\text{erf}(x) = 2/\sqrt{\pi} \int_0^x \exp(-t^2)dt$
<code>erfc</code>	<code>erfc</code>	Complementary error function, $\text{erfc}(x) = 2/\sqrt{\pi} \int_x^\infty \exp(-t^2)dt$
<code>erfcx</code>	<code>erfcx</code>	Scaled complementary error function, $\text{erfcx}(x) = \exp(x^2)\text{erfc}(x)$
<code>error</code>	<code>error</code>	Print error message and abort
<code>eval</code>	<code>execstr</code>	Evaluate string vector with Scilab expressions or statements
<code>exist(a)</code>	<code>exists(a)</code>	Test if variable <code>a</code> exists
<code>exp</code>	<code>exp</code>	Exponential function
<code>expm</code>	<code>expm</code>	Matrix exponential function
<code>eye</code>	<code>eye</code>	Identity matrix (or its generalization)
<code>fclose</code>	<code>mclose</code>	Close (all) open file(s)
<code>fft</code>	<code>fft</code>	Forward and inverse Fast Fourier Transform
<code>fft2</code>	<code>fft</code>	Forward and inverse Fast Fourier Transform
<code>figure</code>	<code>xset</code>	Set defaults for current graphics window
<code>find</code>	<code>spget</code>	Retrieve entries of a sparse matrix
<code>find(a)</code>	<code>find(a)</code>	Find the indices for which boolean matrix <code>a</code> is true
<code>findstr</code>	<code>strindex</code>	Find starting position(s) of a string in an other string
<code>fix(a)</code>	<code>fix(a)</code>	Rounds the elements of <code>a</code> to the nearest integers towards zero
<code>floor(a)</code>	<code>floor(a)</code>	Rounds the elements of <code>a</code> to the nearest integers $\geq a$
<code>fopen</code>	<code>mopen</code>	Open a file
<code>for</code>	<code>for</code>	Start a loop with a generally known number of repetitions
<code>fprintf</code>	<code>fprintf</code>	Write formatted data to file (like C-language <code>fprintf</code> function)
<code>full</code>	<code>full</code>	Convert sparse to full matrix
<code>function</code>	<code>function</code>	Identify header of function definition
<code>gamma</code>	<code>gamma</code>	Gamma function, $\Gamma(x) = \int_0^\infty t^{x-1} \exp(-t)dt$
<code>gammaln</code>	<code>gammaln</code>	Logarithm of the Gamma function, $\ln(\Gamma(x))$
<code>getfield</code>	<code>getfield</code>	Get a data object from a list
<code>global</code>	<code>global</code>	Define variables as global
<code>help</code>	<code>help</code>	On-line help
<code>hess(M)</code>	<code>hess(M)</code>	Hessenberg form of <code>M</code>
<code>if</code>	<code>if</code>	Start a conditionally executed block of statements
<code>ifft</code>	<code>fft</code>	Forward and inverse Fast Fourier Transform
<code>ifft2</code>	<code>fft</code>	Forward and inverse Fast Fourier Transform
<code>imag</code>	<code>imag</code>	Imaginary part of a matrix
<code>input</code>	<code>input</code>	Prompt for user (keyboard) input



Matlab	Scilab	
<code>int16(a)</code>	<code>int16(a)</code>	Convert <b>a</b> to 16-bit signed integer
<code>int32(a)</code>	<code>int32(a)</code>	Convert <b>a</b> to 32-bit signed integer
<code>int8(a)</code>	<code>int8(a)</code>	Convert <b>a</b> to 8-bit signed integer
<code>intersect</code>	<code>intersect</code>	Returns elements common to two vectors
<code>inv(M)</code>	<code>inv(M)</code>	Inverse of matrix <b>M</b>
<code>isempty(a)</code>	<code>a==[]</code>	Check if variable <b>a</b> is empty
<code>isglobal(a)</code>	<code>isglobal(a)</code>	Test if <b>a</b> is a global variable
<code>isinf(a)</code>	<code>isinf(a)</code>	Test if <b>a</b> is infinite
<code>isnan(a)</code>	<code>isnan(a)</code>	Output boolean vector with entries <b>%t</b> where <b>a</b> is <b>%nan</b>
<code>isreal(a)</code>	<code>isreal(a)</code>	Test if <b>a</b> is real (or if its imaginary part is “small”)
<code>keyboard</code>	<code>pause</code>	Interrupt execution of function or script
<code>length</code>	<code>length</code>	Length of list; product of no. of rows and columns of matrix
<code>linspace</code>	<code>linspace</code>	Create vector with linearly spaced elements
<code>log</code>	<code>log</code>	Natural logarithm
<code>log10</code>	<code>log10</code>	Base-10 logarithm
<code>log2</code>	<code>log2</code>	Base-2 logarithm
<code>logm</code>	<code>logm</code>	Matrix natural logarithm
<code>logspace</code>	<code>logspace</code>	Create vector with logarithmically spaced elements
<code>lookfor</code>	<code>apropos</code>	Keyword search for a function
<code>lower</code>	<code>convstr</code>	Convert string to lower or upper case
<code>max</code>	<code>maxi</code>	Maximum of all elements of a vector or array
<code>max</code>	<code>max</code>	Maximum of all elements of a vector or array
<code>mean</code>	<code>mean</code>	Mean of all elements of a vector or array
<code>median</code>	<code>median</code>	Median of all elements of a vector or array
<code>min</code>	<code>min</code>	Minimum of all elements of a vector or array
<code>min</code>	<code>mini</code>	Minimum of all elements of a vector or array
<code>mod(a,b)</code>	<code>pmodulo(a,b)</code>	$a - b \cdot \text{floor}(a/b)$ if $b \neq 0$ ; remainder of <b>a</b> divided by <b>b</b>
<code>more</code>	<code>lines</code>	Specify number of lines to display and columns/line
<code>nargin</code>	<code>argn</code>	Number of input/output arguments of a function
<code>nargout</code>	<code>argn</code>	Number of input/output arguments of a function
<code>nnz</code>	<code>nnz</code>	Number of nonzero elements of a sparse matrix
<code>norm(M)</code>	<code>norm(M)</code>	Norm of <b>M</b> (matrix or vector)
<code>null(M)</code>	<code>kernel(M)</code>	Nullspace of <b>M</b>
<code>num2str</code>	<code>string</code>	Convert numbers to strings
<code>ones</code>	<code>ones</code>	Matrix of ones
<code>orth(M)</code>	<code>orth(M)</code>	Orthogonal basis for the range of <b>M</b>
<code>pause</code>	<code>halt</code>	Stop execution and wait for a key press

Matlab	Scilab	
<code>pinv</code>	<code>pinv(M)</code>	Pseudoinverse of <b>M</b>
<code>planerot</code>	<code>givens</code>	Given's rotation
<code>prod</code>	<code>prod</code>	Product of the elements of a matrix
<code>qr(M)</code>	<code>qr(M)</code>	QR decomposition of <b>M</b>
<code>rand</code>	<code>rand</code>	Create random numbers with uniform or normal distribution
<code>randn</code>	<code>rand</code>	Create random numbers with uniform or normal distribution
<code>rank(M)</code>	<code>rank(M)</code>	Rank of <b>M</b>
<code>rcond(M)</code>	<code>rcond(M)</code>	Reciprocal of the condition number of <b>M</b> ; L-1 norm
<code>real</code>	<code>real</code>	Real part of a matrix
<code>rem(a,b)</code>	<code>modulo(a,b)</code>	<code>a-b.*fix(a./b)</code> if <code>b~=0</code> ; remainder of <b>a</b> divided by <b>b</b>
<code>reshape</code>	<code>matrix</code>	Reshape a vector or a matrix to a different-size matrix
<code>return</code>	<code>resume</code>	Return from a function or resume execution after a <b>pause</b>
<code>return</code>	<code>return</code>	Return from a function or resume execution after a <b>pause</b>
<code>rmfield</code>	<code>null</code>	Delete an element of a list
<code>round(a)</code>	<code>round(a)</code>	Round the elements of <b>a</b> to the nearest integers
<code>schur(M)</code>	<code>schur(M)</code>	Schur decomposition
<code>select</code>	<code>select</code>	Start a multi-branch block of statements
<code>setfield</code>	<code>setfield</code>	Set a data object in a list
<code>sign(a)</code>	<code>sign(a)</code>	Signum function, $a/ a $ for $a \neq 0$
<code>sin</code>	<code>sin</code>	Sine
<code>sinh</code>	<code>sinh</code>	Hyperbolic sine
<code>size</code>	<code>size</code>	Size/dimensions of a Scilab object
<code>sort(a)</code>	<code>gsort(a)</code>	Sort the elements of <b>a</b>
<code>sortrows(a)</code>	<code>gsort(a)</code>	Sort elements/rows/columns of <b>a</b>
<code>spalloc</code>	<code>spzeros</code>	Sparse zero matrix
<code>sparse</code>	<code>sparse</code>	Create sparse matrix
<code>speye</code>	<code>speye</code>	Sparse identity matrix
<code>spones</code>	<code>spones</code>	Replace non-zero elements in sparse matrix by ones
<code>sprand</code>	<code>sprand</code>	Create sparse random matrix
<code>sqrt(a)</code>	<code>sqrt(a)</code>	$\sqrt{a}$
<code>sqrtm</code>	<code>sqrtm</code>	Matrix square root
<code>sscanf</code>	<code>msscanf</code>	Read variables from a string under format control
<code>std</code>	<code>st_deviation</code>	Standard deviation
<code>strrep</code>	<code>strsubst</code>	Substitute one string for another in a third string
<code>sum</code>	<code>sum</code>	Sum of all elements of a matrix
<code>svd</code>	<code>svd</code>	Singular value decomposition
<code>tan</code>	<code>tan</code>	Tangent

Matlab	Scilab	
<code>tanh</code>	<code>tanh</code>	Hyperbolic tangent
<code>tic</code>	<code>timer()</code>	Ouputs time elapsed since the call to <code>timer()</code>
<code>toc</code>	<code>timer</code>	Ouputs time elapsed since the preceding call to <code>timer()</code>
<code>toeplitz</code>	<code>toeplitz</code>	Toeplitz matrix
<code>trace</code>	<code>trace</code>	Trace (sum of diagonal elements) of a matrix
<code>tril</code>	<code>tril</code>	Extract lower-triangular part of a matrix
<code>triu</code>	<code>triu</code>	Extract upper-triangular part of a matrix
<code>try</code>	<code>errcatch</code>	Trap error
<code>uigetfile</code>	<code>xgetfile</code>	Open dialog box for file selection
<code>uint16(a)</code>	<code>uint16(a)</code>	Convert <code>a</code> to 16-bit unsigned integer
<code>uint32(a)</code>	<code>uint32(a)</code>	Convert <code>a</code> to 32-bit unsigned integer
<code>uint8(a)</code>	<code>uint8(a)</code>	Convert <code>a</code> to 8-bit unsigned integer
<code>union</code>	<code>union(a,b)</code>	Extract the unique common elements of <code>a</code> and <code>b</code>
<code>unique(a)</code>	<code>unique(a)</code>	Return the unique elements of <code>a</code> in ascending order
<code>unix</code>	<code>unix_w</code>	Execute Unix/DOS command; output to Scilab window
<code>upper</code>	<code>convstr</code>	Convert string to lower or upper case
<code>varargin</code>	<code>varargin</code>	Variable number of input arguments for a function
<code>varargout</code>	<code>varargout</code>	Variable number of output arguments for a function
<code>version</code>	<code>getversion</code>	Display version of Scilab
<code>warning</code>	<code>warning</code>	Print warning message
<code>which</code>	<code>whereis</code>	Display name of library containing function
<code>while</code>	<code>while</code>	Start repeated execution of a block while a condition is satisfied
<code>who</code>	<code>who</code>	Displays/outputs names of current variables
<code>whos</code>	<code>whos</code>	Displays/outputs names and specifics of current variables
<code>zeros</code>	<code>zeros</code>	Matrix of zeros

# Index

- arguments
  - named, 55
  - variable-length list of, 56, 87
- ASCII codes, 27
- boolean operators, 35
- boolean variables, 35
- cell array, 6, 25, 26, 39–42
- character string, *see* string
- command-style syntax, 15, 54, 81
- comments, 7
- complex number, 21
- constants, built-in, 10
- continuation of a statement, 8
- convolution, 64, 103
- copy, 5
- date, 16
- directory
  - private, *see* private directory
- DOS, *see* MS-DOS
- end of file, 67
- error trapping, 12, 14, 15, 32, 33, 72, 78, 82
- Fast Fourier Transform, 64–65, 104
- FFT, *see* Fast Fourier Transform
- file, 66–75
  - opening, closing, 66
- flow control, 12
- font, 4
- format
  - loose, 2
  - tight, 2
- function
  - as argument, 89
  - as variable, 89
  - inline, 91, 92
  - library, 90
  - loading, 90
  - overloading, 95
  - private, *see* private function
  - Scilab, 54–77
  - user, 84–97
- help, 5, 6
- integer, 74
- interrupt, *see* **pause**
- Kronecker
  - product, 8
  - division, 8
- last element
  - of vector or matrix, 22, 30
- L<sup>A</sup>T<sub>E</sub>X, 76
- length
  - list, 42
  - numeric matrix, 15
  - string, 25, 26, 30
- library, 98–100
- lines displayed, 4
- list, 39
  - matrix-like typed, 49–50
  - ordinary, 39–43
  - polynomials, 50–53
  - typed, 6, 42–49, 96
- Matlab
  - m-file conversion, 97
  - emulation, 6
- Matlab-mode, 24
- matrix, 21, 22, 24, 107
  - companion, 24, 103

- empty, 24, 102
- Frank, 24
- Hilbert, 24
- identity, 24, 104
- of ones, 24, 105
- of zeros, 24, 107
- polynomial, *see* polynomial matrix, 51
- random, 24, 106
- rational, 51
- reshape a, 106
- sparse, 62
- sparse identity, 62, 106
- sparse random, 62, 106
- Toeplitz, 24, 107
- MS-DOS, 77
- operator
  - binary, 95
  - code, 94, 95
  - overloading, 19, 44, 50, 92–97
  - unary, 94
- overloading
  - operator, *see* operator, overloading
  - variable display, *see* variable, display overloading
- paging, *see* lines displayed
- paste, 5
- path, 81
- polynomial, 50–53, 76
  - division, 51
  - matrix, 51
- private directory, 91
- private function, 91
- program
  - exit, 5
  - interrupt, 5, 15
  - quit, 5
  - resume, 5
- Scilab
  - syntax, 7, 8
- script, 78–83
- search path, 32, 81
- size
  - list, 42
  - string, 25–27
- sorting, 58–59
- startup file, 4, 99
- string
  - concatenation, 27, 92
  - extraction, 28
- structure, 1, 39, 44–47
- termination script, 5
- time, 16
- tuple assignment, 9, 41
- type name, 43, 44, 92, 94
- type of variable, 19, 20
  - boolean matrix, 19
  - boolean sparse matrix, 19
  - compiled function, 19
  - constant, 19
  - function library, 19
  - implied-size vector, 19, 22
  - integer, 20
  - list, 19
  - Matlab sparse matrix, 19
  - matrix-like typed list, 19
  - pointer, 19
  - polynomial matrix, 19
  - rational, 38, 51, 53
  - sparse matrix, 19
  - string, 19
  - typed list, 19
  - un-compiled function, 19
- typed list, *see* list, typed
- UNIX, 77
- variable
  - conversion into an expression, 31
  - display overloading, 93, 94
  - global, 16, 87, 88, 104
  - name, 7, 8, 91
  - type, *see* type, variable
- working directory, 78

workspace, 1, 5, 15–17, 33, 73, 78, 86, 88, 89,  
98–100

## Index of Scilab Functions and Variables

`[]`, 24, 102

`$`, 16, 22, 29, 30, 46, 104

`%asn`, 62

`%io`, 77

`%k`, 62

`%sn`, 62, 103

`a==[]`, 105

`abort`, 5

`abs`, 57, 102

`acos`, 60, 102

`acosh`, 60, 102

`acoshm`, 61

`acosm`, 61

`amell`, 62

`and`, 37, 38, 102

`apropos`, 5, 6, 16, 105

`argn`, 85, 88, 105

`ascii`, 26, 27, 29, 30, 103

`asin`, 60, 102

`asinh`, 60, 102

`asinhm`, 61

`atan`, 59, 60, 102

`atanh`, 60, 102

`atanhm`, 61

`auread`, 68

`auwrite`, 69

`balanc`, 63, 102

`bandwr`, 62

`bdiag`, 63, 103

`besseli`, 62, 103

`besselj`, 62, 103

`besselk`, 62, 103

`bessely`, 62, 103

`bezout`, 51

`blanks`, 29

`bool2s`, 38, 57

`break`, 12, 103

`case`, 12, 103

`ceil`, 57, 103

`chdir`, 81

`chfact`, 62

`chol`, 63, 103

`chsolve`, 62

`clean`, 51, 57, 61

`clear`, 16, 99, 100, 103

`clearglobal`, 16, 103

`cmdndred`, 51

`coeff`, 51

`coffg`, 51

`colcomp`, 63

`colcompr`, 51

`companion`, 24, 103

`cond`, 63, 103

`conj`, 57, 103

`convol`, 64, 103

`convstr`, 26, 89, 105, 107

`cos`, 60, 103

`cosh`, 60, 103

`coshm`, 61

`cosm`, 61

`cotg`, 60, 103

`coth`, 60, 103

`cumprod`, 57, 103

`cumsum`, 57, 103

`date`, 16, 26, 103

`deff`, 91, 92

`degree`, 51

`delip`, 62

`denom`, 51

`derivat`, 51

`det`, 51, 63, 103

`determ`, 51

`detr`, 51

`diag`, 24, 103

`diary`, 69, 76, 103

`diophant`, 51

`disp`, 16, 18, 69, 75, 103

`dispfiles`, 67, 101

`dlgamma`, 62  
`double`, 22, 103  
  
`edit`, 90  
`else`, 12, 103  
`elseif`, 12, 104  
`emptystr`, 26, 29, 30  
`end`, 12, 104  
`endfunction`, 85  
`erf`, 62, 104  
`erfc`, 62, 104  
`erfcx`, 62, 104  
`errcatch`, 12, 14, 15, 107  
`errclear`, 14, 15  
`error`, 85, 104  
`eval`, 32  
`evstr`, 32–34  
`excel2sci`, 68  
`exec`, 79, 81  
`execstr`, 32, 78, 82, 104  
`exists`, 15, 36, 38, 104  
`exit`, 5  
`exp`, 60, 104  
`expm`, 61, 104  
`eye`, 24, 104  
  
`factors`, 51  
`fft`, 64, 104  
`file`, 66, 67, 71, 72, 82, 83  
`fileinfo`, 67  
`find`, 38, 104  
`fix`, 57, 104  
`floor`, 57, 104  
`for`, 12, 104  
`fprintf`, 69, 104  
`fprintfMat`, 69  
`fscanf`, 68  
`fscanfMat`, 68  
`full`, 62, 104  
`fullrf`, 63  
`fullrfk`, 63  
`function`, 85, 104  
  
`gamma`, 62, 104  
`gammaln`, 62, 104  
`gcd`, 51  
`genlib`, 90, 98  
`getd`, 90  
`getdate`, 16  
`getf`, 90, 100, 101  
`getfield`, 40, 104  
`getio`, 67  
`getversion`, 76, 107  
`givens`, 63, 106  
`global`, 16, 81, 86–88, 104  
`grand`, 24  
`grep`, 26, 30  
`gsort`, 26, 57–59, 106  
  
`halt`, 15, 16, 85, 105  
`help`, 5, 6, 16, 104  
`hermit`, 51  
`hess`, 63, 104  
`horner`, 51  
`host`, 76, 77  
`householder`, 63  
`hrmt`, 51  
  
`iconvert`, 22  
`if`, 12, 13, 104  
`imag`, 57, 104  
`input`, 68, 75, 104  
`int16`, 22, 105  
`int32`, 22, 105  
`int8`, 22, 23, 105  
`intersect`, 26, 57, 105  
`inttype`, 16, 20  
`inv`, 51, 63, 105  
`invr`, 51  
`isdef`, 36, 38  
`iserror`, 14, 38  
`isglobal`, 38, 105  
`isinf`, 38, 105  
`isletter`, 27  
`ismember`, 39  
`isnan`, 38, 105  
`isreal`, 38, 105

- kernel, 63, 105
- lcm, 51
- lcmdiag, 51
- ldiv, 51
- length, 6, 26, 40, 42, 105
- lex\_sort, 57
- lines, 76, 105
- linsolve, 63
- linspace, 57, 105
- list, 19, 39, 40
- load, 68, 73, 74, 90, 99
- loadwave, 68
- log, 60, 105
- log10, 60, 105
- log2, 60, 105
- logm, 61, 105
- logspace, 57, 105
- lstcat, 40
- lufact, 62
- luget, 62
- lusolve, 62
- matrix, 70, 106
- max, 56, 57, 105
- maxi, 57, 105
- mclearerr, 67
- mclose, 66, 67, 69, 101, 104
- mean, 57, 105
- median, 57, 105
- meof, 67
- mfft, 64
- mfprintf, 69
- mfscanf, 68
- mget, 68, 74, 75
- mgeti, 68, 74
- mgetl, 67–70
- mgetstr, 68
- min, 57, 105
- mini, 57, 105
- mlist, 19, 39, 40, 49
- mode, 78–81, 85, 103
- modulo, 57, 106
- mopen, 66–68, 71, 73, 75, 104
- mprintf, 69
- mput, 69, 74
- mputl, 67, 69–71
- mputstr, 69
- mscanf, 68
- MSDOS, 38, 77
- mseek, 67
- msprintf, 26
- msscanf, 26, 31, 106
- mtell, 67
- mtlb\_cell, 6
- mtlb\_diff, 6
- mtlb\_load, 68
- mtlb\_mode, 24, 38
- mtlb\_save, 69
- newest, 67
- nnz, 62, 105
- norm, 63, 105
- null, 40, 41, 106
- ones, 24, 36, 105
- or, 37, 38, 102
- orth, 63, 105
- otherwise, 12
- part, 26, 29, 82
- pause, 5, 15–17, 85, 88, 105
- pdiv, 51
- pinv, 63, 106
- pmodulo, 57, 105
- pol2str, 26, 51
- pol2tex, 76
- polar, 63
- poly, 50
- print, 69
- printf, 69
- prod, 57, 106
- qr, 63, 106
- quit, 5
- rand, 24, 106
- range, 63
- rank, 63, 106



- rcond, 63, 106
- read, 68, 70, 71
- read4b, 68
- readb, 68
- readc\_, 68
- real, 57, 106
- reshape, 70
- residu, 51
- resume, 5, 18, 85, 88, 106
- return, 5, 16–18, 82, 85, 88, 106
- roots, 51
- round, 57, 106
- rowcompr, 51
  
- save, 68, 73, 74
- savewave, 69
- scanf\_conversion, 31
- schur, 63, 106
- sci2exp, 31
- sci2expr, 26
- select, 12, 106
- setfield, 40, 106
- sfact, 51
- sign, 57, 106
- simp, 51
- simp\_mode, 38, 51, 52
- sin, 60, 106
- sinh, 60, 106
- sinhm, 61
- simm, 61
- size, 25, 26, 40, 42, 106
- spaninter, 63
- spanplus, 63
- sparse, 62, 106
- spchol, 62
- spec, 63, 103
- speye, 62, 106
- spget, 62, 104
- spones, 62, 106
- sprand, 62, 106
- spzeros, 62, 106
- sqrt, 57, 106
- sqrtm, 61, 106
- st\_deviation, 57, 106
  
- stacksize, 76
- strcat, 26, 28
- strindex, 26, 30, 104
- string, 26, 105
- stripblanks, 26, 29, 103
- strsubst, 26, 106
- sum, 57, 106
- sva, 63
- svd, 63, 106
- sylm, 24, 51
  
- tan, 60, 106
- tanh, 60, 107
- tanhm, 61
- tanm, 61
- testmatrix, 24
- texprint, 76
- timer, 16, 76, 107
- tlist, 19, 39, 40, 43, 46
- toeplitz, 24, 107
- trace, 63, 107
- tril, 107
- triu, 107
- type, 16, 19, 20, 91, 92
- typename, 20
- typeof, 16, 19, 20, 23, 31, 89, 92
  
- uint16, 22, 107
- uint32, 22, 107
- uint8, 20, 22, 107
- union, 26, 57, 107
- unique, 26, 57, 107
- unix, 76
- unix\_g, 76, 77
- unix\_s, 76
- unix\_w, 76, 98, 107
- unix\_x, 76
  
- varargin, 56, 85, 107
- varargout, 85, 107
  
- warning, 85, 107
- wavread, 68
- wavwrite, 69
- where, 85, 103

*whereami*, 85, 103  
*whereis*, 16, 85, 107  
*while*, 12, 107  
*who*, 16, 20, 99, 107  
*whos*, 16, 20, 107  
*writb*, 69  
*write*, 69–73, 77  
*write4b*, 69  
  
*xgetfile*, 67, 75, 107  
*xset*, 104  
  
*zeros*, 24, 36, 107

## Index of Matlab Functions and Variables

*[]*, 102  
  
*abs*, 102  
*acos*, 102  
*acosh*, 102  
*all*, 37, 102  
*any*, 37, 38, 102  
*asin*, 102  
*asinh*, 102  
*atan*, 59, 102  
*atan2*, 59, 102  
*atanh*, 102  
  
*balance*, 102  
*besseli*, 103  
*besselj*, 103  
*besselk*, 103  
*bessely*, 103  
*blanks*, 29  
*break*, 12, 103  
  
*case*, 12, 103  
*catch*, 12, 14  
*ceil*, 103  
*cell*, 6, 30  
*char*, 27, 103  
*chol*, 103  
*clear*, 103  
  
*clear global*, 103  
*compan*, 103  
*complex*, 21  
*cond*, 103  
*conj*, 103  
*conv*, 103  
*cos*, 103  
*cosh*, 103  
*cot*, 103  
*coth*, 103  
*cumprod*, 103  
*cumsum*, 103  
  
*date*, 103  
*dbstack*, 103  
*deal*, 9  
*deblank*, 29, 103  
*det*, 103  
*diag*, 103  
*diary*, 76, 103  
*diff*, 6  
*disp*, 103  
*double*, 27, 103  
  
*echo*, 103  
*eig*, 103  
*ellipj*, 103  
*else*, 12, 103  
*elseif*, 12, 104  
*end*, 22  
*end [loop]*, 12, 104  
*end [matrix]*, 46, 104  
*erf*, 104  
*erfc*, 104  
*erfcx*, 104  
*error*, 104  
*eval*, 32, 104  
*evalin*, 33  
*exist*, 15, 104  
*exp*, 104  
*expm*, 104  
*eye*, 104  
  
*fclose*, 104

*feval*, 89  
*fft*, 64, 104  
*fft2*, 64, 104  
*figure*, 104  
*find*, 104  
*findstr*, 104  
*finish*, 5  
*fix*, 104  
*floor*, 104  
*fopen*, 66, 75, 104  
*for*, 12, 104  
*fprintf*, 104  
*full*, 104  
*function*, 104  
*funm*, 61  
  
*gamma*, 104  
*gammaln*, 104  
*getfield*, 104  
*global*, 104  
  
*help*, 5, 104  
*hess*, 104  
  
*if*, 12, 104  
*ifft*, 64, 104  
*ifft2*, 64, 104  
*imag*, 104  
*input*, 75, 104  
*int16*, 105  
*int32*, 105  
*int8*, 105  
*intersect*, 105  
*inv*, 105  
*isa*, 20  
*iscell*, 19  
*ischar*, 19  
*isempty*, 105  
*isglobal*, 105  
*isinf*, 105  
*isletter*, 27  
*ismember*, 31, 38  
*isnan*, 105  
*isnumeric*, 19  
  
*isreal*, 105  
*issparse*, 19  
*isstruct*, 19  
  
*keyboard*, 17, 88, 105  
  
*length*, 105  
*linspace*, 105  
*load*, 74  
*log*, 105  
*log10*, 105  
*log2*, 105  
*logical*, 35, 36  
*logm*, 105  
*logspace*, 105  
*lookfor*, 5, 105  
*lower*, 105  
  
*max*, 57, 105  
*mean*, 105  
*median*, 105  
*min*, 105  
*mod*, 105  
*more*, 105  
  
*nargin*, 105  
*nargout*, 105  
*nnz*, 105  
*norm*, 105  
*null*, 105  
*num2str*, 105  
  
*ones*, 105  
*orth*, 105  
*otherwise*, 12  
  
*pause*, 105  
*pinv*, 106  
*planerot*, 106  
*prod*, 106  
  
*qr*, 106  
  
*rand*, 106  
*randn*, 106  
*rank*, 106

*rcond*, 106  
*real*, 106  
*rem*, 106  
*reshape*, 106  
*return*, 106  
*rmfield*, 106  
*roots*, 52  
*round*, 106

*save*, 73  
*schur*, 106  
*select*, 106  
*setfield*, 106  
*sign*, 106  
*sin*, 106  
*sinh*, 106  
*size*, 26, 50, 106  
*sort*, 58, 106  
*sortrows*, 58, 106  
*spalloc*, 106  
*sparse*, 106  
*speye*, 106  
*spones*, 106  
*sprand*, 106  
*sqrt*, 106  
*sqrtn*, 106  
*sscanf*, 31, 106  
*std*, 106  
*strrep*, 106  
*sum*, 106  
*svd*, 106  
*switch*, 12

*tan*, 106  
*tanh*, 107  
*tic*, 107  
*toc*, 107  
*toeplitz*, 107  
*trace*, 107  
*tril*, 107  
*triu*, 107  
*try*, 12, 14, 107  
*type*, 53  
*typeof*, 50, 53

*uigetfile*, 75, 107  
*uint16*, 107  
*uint32*, 107  
*uint8*, 20, 107  
*union*, 107  
*unique*, 107  
*unix*, 107  
*upper*, 107

*varargin*, 107  
*varargout*, 107  
*version*, 107

*warning*, 107  
*which*, 107  
*while*, 12, 107  
*who*, 107  
*whos*, 107

*zeros*, 107